AD-A162 794 HELICOPTER PERFORMANCE COMPUTER PROGRAMS FOR HP-41 HAND-HELD COMPUTER(U) NAVAL POSTGRADUATE SCHOOL MONTEREY CA D M LAYTON DEC 85 1/1

UNCLASSIFIED F/G 1/3 NL





MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A



HP-41

Helicopter Programs

Prof D.M.Layton

December 1985

HELICOPTER PERFORMANCE

COMPUTER PROGRAMS

for

HP-41 Hand-held Computer

Prof Donald M. Layton
Department of Aeronautics
Naval Postgraduate School
Monterey, California

TABLE OF CONTENTS



INTRODUCTION

These programs present the user of the HP-41 hand-held, programable calculator with a series of self-prompting, alphanumeric programs that can be used with acceptable results to compute and evaluate helicopter performance. Most of the programs are stuctured so as to be more expedient when doing preliminary helicopter design where intermediate values are of less a concern than would be the case when studying detailed performance.

By presenting only the principal, 'bottom line', values, the programs are shorter than if all of the intermediate values were displayed. However, with the establishment of a standard storage register routine (see Tables I and II), any intermediate value may be readily located and displayed by using the RCL routine.

The standard storage registers also permit direct loading of data from a mass storage unit such as the HP 82161A Digital Cassette Drive.

The basic equations and assumptions inherent in these programs are from "Helicopter Performance" (reference 1) and "Helicopter Design Manual (reference 2), both by Professor Donald M. Layton. These materials are used regularly in helicopter performance and design courses at the Naval Postgraduate School, Monterey, California.

The programs have been written in a consolidated manner, thus reducing the need to load numerous individual subroutines in order to run a single program. Because of this, many of the programs are quite large and use a considerable amount of the available space in the resident memory of the HP-41CV. If a HP-41C is to be used a Quad Memory Module is required. The option to operate a specific subroutine is maintained for most of the programs. For example, if one desired to ascertain the induced velocity of the main rotor, once data parameters are loaded, that information could be gained by executing that subroutine, e.g., XEQ VI.

Table III is a matrix of program input parameters and programs. Until the user of these programs becomes quite conversant with the programs inputs, it would be wise to check Table III prior to executing a program. For example, the POWER program requires an input of Equivalent Flat Plate Area, and if the drag information is available only in the form of Equivalent Flat Plate Loading, the data must be converted prior inputing FF.

In order to facilitate partial changes of data input sets, most data parameters recall the existing data value prior to asking the prompt question. This permits the maintenance of the same data value by either re-entering the value or pressing R/S.

Each program is documented in essentially the same manner. Listed alphabetically, each program documentation begins with the name of the program (as it appears in the execute statement) and a short phrase that describes the program. This is followed by the introduction which states the purpose of the program, explains the applicability of the program, and describes any assumptions inherent in the program.

For a program which requires the loading of additional programs for execution, an entry following the introduction indicates the required programs.. This is, in turn, followed by a listing of the equations used in the program, together with the source of each equation.

An example problem, executed with detailed step-by-step instructions, is furnished with each program. These instructions indicate the inputs required to be made by the user, functions that the user must execute and displays as they would appear on the HP-41. For those program where multiple internal branching occurs, either a representative branch is shown or all branches are demonstrated. For the latter case, indices, such as A or B, follow the appropriate step number.

Within the detailed instructions, the user will be prompted for two types of inputs. The first, and more predominant type, occurs when a numerical input is required by the program. This type of prompt will have the data variable followed by an equality sign and a question mark, e.g. VF=?. The other type of prompt occurs at branch points where a question is asked, the answer to which is either Yes or No. These prompts consist of a symbol, word or short phrase followed only by a question mark, e.g., NEED DATA?. If the answer to this type of prompt is Yes, enter 1, and if the answer is No, enter 0.

The last documentation of each program is a complete listing of the program. This permits the user to see how and why the program arrives at its solutions. It also allows for editing, should adjustments be desired. Although the programs have been written in a straight-forward manner, some combination of constants have been used. While this shortens the program, it does add confusion to any editing of the program.

The backbone of these programs for performance determination is the program POWER. In may instances, this program is required to be loaded and run in conjunction with another program. For example, the highspeed effects program, HSE, makes power corrections to the basic POWER program.

A program. FLITE, that is somewhat similar to POWER is also included for two reasons. First of all, there are provisions in this program for non-rectangular rotor blades. This capability was not incorporated in POWER due to the size of the program. Secondly, FLITE can be used for partial, en-route solutions such as main rotor only, tail rotor only, or individual subroutines.

These programs are all written in British Units. If it is desired to convert to SI Units it is only necessary to modify the aerodynamic parameters (such as density) and to make the proper conversionfor power. For example, if the density for Profile Power is input as $kg\text{-}sec^2/m^4$ and all of the measurements are in meters, to obtain horsepower from kg-m/sec, divide by 76, vice 550 as with British units. The author has available a set of these programs in SI Units.

These programs have been prepared, revised, modified and edited over a several years with the assistance of students in the Aeronautics Programs at the Naval Postgraduate School. Although their contributions are deeply appreciated, the list of these students has now grown so long as to make a complete acknowledgement impractical.

TABLE I
STANDARD STORAGE REGISTER UTILIZATION

Storage Register	Stored Quantity
00	R - Main rotor radius (ft)
01	c - Main rotor equivalent chord (ft)
02	RV - Main rotor rotational velocity (Rad/sec)
03	C_{d_o} - Main rotor profile drag coefficient
04	b - Number of main rotor blades
05	R<TR> - Tail rotor radius (ft)
06	c<TR> -Tail rotor equivalent chord (ft)
07	RV<TR> - Tail rotor rotational velocity (rad/sec)
08	C_{d_o} <TR> - Tail rotor profile drag coefficient
09	b<TR> - Number of tail rotor blades
10	L - Length of tail boom (ft)
11	W - Gross weight (lbs)
12	FF - For. flight Equivalent Flat Plate Area (ft^2)
13	FV - Vert. Equivalent Flat Plate Area (ft^2)
14	RTR HT - Main rotor height above skid/wheel (ft)
15	c_o - Root chord of main rotor (ft)
16	c_1 - Tip chord of main rotor (ft)
17	a CHORD - Main rotor span fraction (for taper)
18	VF - Forward velocity of aircraft (ft/sec) (Entered in kts, stored in ft/sec)
19	VV(FPM) - Vertical velocity of aircraft (ft/sec) (Entered in ft/min, stored in ft/sec))
20	PA/DA - Pressure or Density Altitude (ft)

21 TEMP<F> - Temperature in ^{o}Rankine (Entered in ^{o}F, stored in ^{o}R))

22 Density (ρ) - Ambient density (slugs/ft^3)

23 Area - Main rotor disc area (ft^2)

24 Area<TR> - Tail rotor disc area (ft^2)

25 SKID HT - Height of skids (wheels) above the ground (ft)

26 h/D ratio - Ratio of rotor height to diameter

27 V_T - Main rotor tip velocity (ft/sec)

28 T_T<TR> - Tail rotor tip velocity (ft/sec)

29 C_T - Main rotor coefficient of thrust

30 C_T<TR> - Tail rotor coefficient of thrust

31 VI - Main rotor induced velocity (ft/sec)

32 VI<TR> - Tail rotor induced velocity (ft/sec)

33 B - Main rotor tip loss factor

34 B<TR> - Tail rotor tip loss factor

35 PI - Main rotor induced power (includes tip loss and ground effect) (SHP)

36 PO - Main rotor profile power (SHP)

37 PP - Main rotor parasite power

38 PC - Main rotor climb power (SHP)

39 PT<MR> - Main rotor total power (SHP)

40 T<TR> - Tail rotor thrust (lbs)

41 PI<TR> - Tail rotor induced power (includes tip loss) (SHP)

42 PO<TR> - Tail rotor profile power (SHP)

43 PT<TR> - Tail rotor total power (SHP)

44 PT<AC> - Aircraft total power required (SHP)

45 PI/PI_{OGE} - Ground effect induced power ratio

TABLE II

STANDARD STORAGE REGISTER UTILIZATION

Stored Quantity	Storage Register
a CHORD - Main rotor span fraction	17
Area - Main rotor disc area (ft^2)	23
Area<TR> - Tail rotor disc area (ft^2)	24
b - Number of main rotor blades	04
b<TR> - Number of tail rotor blades	09
B - Main rotor tip loss factor	33
B<TR> -Tail rotor tip loss factor	34
c - Main rotor equivalent chord (ft)	01
c<TR> - Tail rotor chord (ft)	06
C_{d_o} - Main rotor average profile drag coefficient	03
C_{d_o}<TR> - Tail rotor profile drag coefficient	08
c_0 - Main rotor root chord (ft)	15
c_1 - Main rotor tip chord (ft)	16
CT - Main rotor coefficient of thrust	29
CT<TR> - Tail rotor coefficient of thrust	30
Density (ρ) - Ambient density ($slugs/ft^3$)	22
FF - For. Flight Equivalent Flat Plate Area (ft^2)	12
FV - Vert. Equivalent Flat Plate Area (ft^2)	13
h/D - Rotor height to rotor diameter ratio	26
L - Tail boom length (ft)	10
PA/DA - Pressure or density altitude (ft)	20
PC - Main rotor climb power (SHP)	38

PI - Main rotor induced power (includes tip loss and ground effect) (SHP) 35

PI<TR> - Tail rotor induced power (includes tip loss)(SHP) 41

PI/PI_{OGE} - Induced power ground effect ratio 45

PO - Main rotor profile power (SHP) 36

PO<TR> - Tail rotor profile power (SHP) 42

PP - Main rotor parasite power (SHP) 36

PT<AC> - Total aircraft power required (SHP) 44

PT<MR> - Main rotor total power required (SHP) 39

PT<TR> - Tail rotor total power required (SHP) 42

R - Main rotor radius (ft) 00

R<TR> - Tail rotor radius (ft) 05

RTR HT - Main rotor height abopve skids/wheels (ft) 14

RV - Main rotor rotational velocity (rad/sec) 02

RV<TR> - Tail rotor rotational velocity (rad/sec) 07

SKID HT - Height of skids/wheels above ground (ft) 25

T<TR> - Tail rotor thrust (lbs) 40

TEMP<F> - Temperature (entered in ^{o}F, stored in ^{o}R) 21

V_T - Main rotor tip velocity (ft/sec) 27

V_T<TR> - Tail rotor tip velocity (ft/sec) 28

VF - Forward velocity (enter in kts, stored in ft/sec) 18

VI - Main rotor induced velocity (ft/sec) 31

VI<TR> - Tail rotor induced power (ft/sec) 32

VV<FPM> - Vertical velocity (Enter in ft/min, stored in ft/sec) 19

W - Gross weight (lbs) 11

TABLE III
PROGRAM INPUT REQUIREMENTS

	AUTO	CLG	FLITE	FUEL	HSE	POWER	RC	VE	VMR	WT	
a CHORD			x								ft
ALPHA HAT				x							
b		x		x			x			x	
b<TR>		x		x							
BETA HAT				x							
c		x		x						x	ft
CARGO										x	lbs
CL <RAD>					x						
c<TR>			x			x					ft
C_{d_o}			x			x					
C_{d_o} <TR>			x			x					
c_o			x								ft
c_l			x								ft
ENG WT										x	lbs
FF			x		x	x					ft^2
FUEL WT										x	lbs
FV			x		x	x					ft^2
INCR (Velocity)								x	x		kts (ft/sec)*
L			x			x					ft
NENG				x							
PA/DA			x			x					ft

* Stored Quantity

	AUTO	CLG	FLITE	FUEL	HSE	POWER	RC	VE	VMR	WT	
PEOPLE										x	
PERSON WT										x	lb
PSHP									x		shp
PT<AC>										x	shp
RSHP<SSL>		x					x				shp
R	x		x		x	x				x	ft
R<TR>			x	x							ft
RTR<HT>			x			x					ft
RV	x		x		x	x				x	rad/sec
RV<TR>			x			x					rad/sec
SKID HT			x			x					ft
T<TR>											lbsf
TEMP<F>			x	x							$^{o}_{F}$ (^{o}R)*
TRAN WT											lbs
TWIST					x						deg
VF	x		x	x	x	x	x				kts (ft/sec)*
V-START								x	x		kts (ft/sec)*
V-STOP								x	x		kts (ft/sec)*
VV			x			x					ft/min (ft/sec)*
W	x		x		x	x				x	lbs

* Stored Quantity

AUTO

(Autorotation)

Introduction: This program computes approximations for both mimimum rate of descent vertical autorotation and minimum descent rate forward autorotation. It uses the standard input registers, and if data is required, all parameters are requested, even though only a few are needed.

ADDITIONAL PROGRAMS REQUIRED: None

Equations

$$\bar{C}_L = (3K_2/K_1)^{\frac{1}{2}}$$ Ref 1, Eqn 6-14

$$\bar{C}_d = K_1\bar{C}_L^{\,2} + K_2$$ Ref 1, Eqn 6-15

$$\bar{F} = \frac{(C_L^{\,3}/C_d^{\,2})\cdot\sigma}{4}$$ Ref 1, Eqn 6-8

$$V_v = \left[\frac{W}{2\cdot\rho\cdot A_D\cdot\bar{f}}\right]^{\frac{1}{2}}$$ Ref 1, Eqn 6-11

$$\bar{f} = \frac{\bar{F}}{(1+\bar{F})^2} \qquad (0<\bar{F}<1)$$ Ref 1, Eqn 6-9

$$\bar{f} = \frac{(2\bar{F} - \sqrt{3\bar{F}})}{(4\bar{F}-3)} \qquad (\bar{F}>1)$$ Ref 1, Eqn 6-10

$$V_{f(\text{min ROD})} = 0.00867 \cdot R \cdot RPM$$ Ref 1, Eqn 6-17

$$V_{v(\text{min ROD})} = 0.251 \cdot R \cdot RPM$$ Ref 1, Eqn 6-18

$$d_{(\text{hor glide})} = \frac{h}{\tan\gamma}$$ Ref 1, Eqn 6-19

$$\gamma = \arcsin\frac{V_v}{V_f} \approx 16.6^\circ$$ Ref 1, Eqn 6-20

where

$\bar{C}_L$ is the average coefficient of lift

$\bar{C}_d$ is the average coefficient of drag

K_1 is a real number coefficient called the lift coefficient multiplier in drag coefficient terms

K_2 is a real number coefficient equal to C_{d_o}

V_v is the vertical velocity in a vertical autorotation (ft/min)

A_D is the area of the rotor disc (ft^2)

σ is the solidity of the main rotor system

ρ is the density of the air $\left[\frac{lb\text{-}sec^2}{ft^4}\right]$

h is the height of the rotor system above the ground (ft)

RPM is the rotational velocity of the main rotor system in revolutions/minute

$\bar{F}$ is a non-dimensional coefficient

$\bar{f}$ is a non-dimensional coefficient

W is the weight of the helicopter (lbs)

R is the radius of the rotor system (ft)

γ is the descent angle for minimum descent rate (degrees)

=

VV=	vertical velocity in a vertical autorotation (ft/min)
VF(MIN.R.O.D.)=	forward autorotative flight velocity for minimum autorotative rate of descent (kts)
VV(MIN.R.O.D.)=	vertical autorotative velocity (ft/min) at the forward autorotative flight velocity for minimum autorotative rate of descent
d(HOR.GLIDE)=	horizontal distance travelled on the ground at the forward autorotative flight velocity for minimum rate of descent (ft)

SIZE 060

INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1. Initialize the program		XEQ AUTO	NEED DATA?
2 Answer 1 for yes, 0 for no	1	R/S	W=?
3. Input weight (lbs)	20000	R/S	RV=?
4. Input main rotor rotational velocity (rad/sec)	27	R/S	b=?
5. Input number of main rotor blades	4	R/S	c=?
6. Input main rotor chord (ft)	1.75	R/S	CdO=?
7. Input main rotor drag coefficient	.008	R/S	R=?
8. Input main rotor radius (ft)	26.8	R/S	FF=?
9. Input forward flat plate area (sq ft)	25.7	R/S	FV=?
10. Input vertical flat plate area (sq ft)	30.8	R/S	RV(TR)=?
11. Input tail rotor rotational velocity (rad/sec)	124.6	R/S	b(TR)=?
12. Input number of tail rotor blades	4	R/S	c(TR)=?
13. Input tail rotor chord (ft)	.81	R/S	CdO(TR)=?
14. Input tail rotor chord (ft)	.008	R/S	R(TR)=?
15. Input tail rotor radius (ft)	5.5	R/S	L(TAIL)=?

INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
16. Input length of tail (ft)	31.5	R/S	RTR HT=?
17. Input rotor height above skids (ft)	10	R/S	SKID HT=?
18. Input skid height above ground (ft)	4000	R/S	LCM=?
19. Input lift coefficient multiplier (Kl)	.004	R/S	VV=
20. Outputs vertical velocity in vertical autorotative (ft/min)	0	R/S	VF(MIN ROD)=
21. Outputs flight velocity for minimum rate of descent (kts)	59.9	R/S	VV(MIN ROD)=
22. Outputs vertical velocity at minimum rate of descent (ft/min)	1816.3	R/S	d(HOR GLIDE)=
23. Outputs horizontal distance at minimum rate of descent (ft)	6092.7		

AUTO

```
01♦LBL "AUTO"
02 FIX 1
03 CF 02
04 "NEED DATA?"
05 PROMPT
06 X=0?
07 GTO "PGM"
08 RCL 11
09 "W=?"
10 PROMPT
11 STO 11
12 RCL 02
13 "RV=?"
14 PROMPT
15 STO 02
16 RCL 04
17 "b=?"
18 PROMPT
19 STO 04
20 RCL 01
21 "c=?"
22 PROMPT
23 STO 01
24 RCL 03
25 "CdO=?"
26 PROMPT
27 STO 03
28 RCL 00
29 "R=?"
30 PROMPT
31 STO 00
32 RCL 12
33 "FF=?"
34 PROMPT
35 STO 12
36 RCL 13
37 "FV=?"
38 PROMPT
39 STO 13
40 RCL 07
41 "RV(TR)=?"
42 PROMPT
43 STO 07
44 RCL 09
45 "b(TR)=?"
46 PROMPT
47 STO 09
48 RCL 06
49 "c(TR)=?"
50 PPOMPT
51 STO 06
52 RCL 08
53 "CdO(TR)=?"
54 PROMPT
55 8
56 RCL 05
57 "R(TR)=?"
58 PROMPT
59 STO 05
60 RCL 10
61 "L(TAIL)=?"
62 PROMPT
63 STO 10
64 RCL 14
65 "RTR HT=?"
66 PROMPT
67 STO 14
68 "SKID HT=?"
69 PROMPT
70 STO 25
71♦LBL "PGM"
72 "LCM=?"
73 PROMPT
74 STO 46
75♦LBL "A1"
76 RCL 46
77 1/X
78 3
79 *
80 RCL 03
81 *
82 SQRT
83 STO 47
84 X↑2
85 RCL 46
86 *
87 RCL 03
88 +
89 X↑2
90 1/X
91 RCL 47
92 3
93 Y↑X
94 *
95 RCL 04
96 *
97 RCL 01
98 *
99 RCL 00
100 /
101 PI
102 /
103 4
104 /
105 STO 48
106 X<0?
107 GTO 01
108 1
109 -
110 "X>0?"
111 GTO 02
112 2
113 +
114 X↑2
115 1/X
116 RCL 48
117 *
118 GTO 03
119♦LBL 02
120 RCL 48
121 3
122 *
123 SQRT
124 CHS
125 RCL 48
126 2
127 *
128 +
129 RCL 48
130 4
131 *
132 3
133 -
134 /
135 RCL 48
136 6.875 E-6
137 *
138 1
139 +
140 4.2561
141 Y↑X
142 .0023769
143 *
144 *
145 RCL 00
146 X↑2
147 *
148 PI
149 *
150 2
151 *
152 1/X
153 RCL 11
154 *
155 SQRT
156 60
157 *
158 "VV="
159 PROMPT
160 VIEW X
161 STOP
162 RCL 02
163 RCL 00
164 *
165 .082798
166 *
167 "VF(MIN,R,O,D)="
168 PROMPT
169 VIEW X
170 STOP
171 30.3158
172 *
173 "VV(MIN,R,O,D)="
174 PROMPT
175 VIEW X
176 STOP
177 "ALT(FT)=?"
178 STO 25
179 .29811
180 /
181 "d(HOR.GLIDE)="
182 PROMPT
183 VIEW X
184 STOP
185 GTO 04
186♦LBL 01
187 "F="
188 ARCL X
189 AVIEW
190♦LBL 04
191 END
```

CLG

Helicopter Hover, Service and Combat Ceilings

Introduction: This program will determine the hover, service and combat ceilings for a helicopter. It is run in conjunction with POWER (which must be loaded up through and including step 19 of the step-by-step instructions). For the three different ceilings, rates of climb of 0, 100, and 500 feet per minute are imposed upon the aircraft. The power is computed for the selected airspeed (0 for hover) and compared with the maximum rotor shaft horsepower available. There are no new equations utilized. CLG utilizes the same storage registers as POWER and additionally those listed below.

Additional Programs Required: POWER

Additional Storage Registers:

Storage Registers	Quantity Stored
48	scratch
58	RSHP(SSL) - maximum rotor shaft horsepower available at standard sea level conditions (SHP)

CLG

	INSTRUCTION	INPUT	FUNCTION	SIZE 060 DISPLAY
1.	Initialize program			XEQ CLG
2.	Reminder flashes			*LOAD POWER*
3.	Reminder flashes			HOVER CLG?
4.	Do you want hover ceiling? Input 1 for Yes	1	R/S	RSHP(SSL)=?
5.	Input SSL rotor shaft horsepower available (SHP)	2500	R/S	HOVER CLG= 3,770
6.	Ouput hover ceiling (feet)			

- or to get service ceiling -

2.	Do you want hover ceiling? Input 0 for No	0	R/S	SERVICE CLG?
3.	Do you want service ceiling? Input 1 for yes	1	R/S	VF=?
4.	Input forward velocity (kts)	90	R/S	RSHP(SSL)=?
5.	Input SSL rotor shaft horsepower available (SHP)	2500	R/S	SERVICE CLG= 17543
6.	Output service ceiling (feet)			

- Or to get combat ceiling

	INSTRUCTION	INPUT	FUNCTION	DISPLAY
3.	Do you want service ceiling? Input o for No	0	R/S	COMBAT CLG?
4.	Do you want combat ceiling? Input 1 for Yes	1	R/S	VF=?
5.	Input forward velocity (kts)	90	R/S	RSHP(SSL)=?
6.	Input SSL rotor shaft horsepower available (SHP)	2500	R/S	COMBAT CLG= 15758
7.	Ouput combat ceiling (feet)			

CLG

```
01*LBL "CLG"
02 "*LOAD POWER*"
03 AVIEW
04 PSE
05 SF 03
06 CF 06
07 CF 07
08 CF 08
09 FIX 0
10*LBL 00
11 "HOVER CLG?"
12 PROMPT
13 X=0?
14 GTO 01
15 SF 06
16 0
17 STO 18
18 STO 19
19 GTO 04
20*LBL 01
21 "SERVICE CLG?"
22 PROMPT
23 X=0?
24 GTO 02
25 SF 07
26 GTO 03
27*LBL 02
28 "COMBAT CLG?"
29 PROMPT
30 X=0?
31 GTO 00
32 SF 08
33*LBL 03
34 "VF=?"
35 PROMPT
36 1.68389
37 *
38 STO 18
39 FS? 07
40 100
41 FS? 08
42 500
43 ENTER↑
44 60
45 /
46 STO 19
47*LBL 04
48 "RSHP(SSL)=?"
49 PROMPT
50 STO 58
51 6561
52 STO 48
53 0
54 STO 20
55*LBL "DEL"
56 RCL 20
57 6.875 E-6
58 *
59 CHS
60 1
61 +
62 STO 55
63 5.2561
64 Y↑X
65 STO 47
66 RCL 55
67 518.68
68 *
69 STO 21
70 RCL 47
71 RCL 55
72 /
73 .0023769
74 *
75 STO 22
76 CF 02
77 XEQ "CT"
78 RCL 55
79 SQRT
80 RCL 47
81 *
82 RCL 58
83 *
84 RCL 44
85 X>Y?
86 GTO 05
87 RCL 48
88 ST+ 20
89 GTO "DEL"
90*LBL 05
91 RCL 48
92 ST- 20
93 3
94 /
95 STO 48
96 1
97 X=Y?
98 GTO "ANS"
99 GTO "DEL"
100*LBL "ANS"
101 RCL 20
102 1
103 -
104 FS? 06
105 "HOVER CLG="
106 FS? 07
107 "SERVICE CLG="
108 FS? 08
109 "COMBAT CLG="
110 ARCL X
111 AVIEW
112 STOP
113 END
```

FLITE
Basic Power Requirements

Introduction: The user inputs the basic geometric and flight parameters of the helicpter and the program determines the steady state power requirements to maintain a given flight condition. In the FLITE program, high speed effects are neglected. The program will determine the power required for the main rotor only, or for the main rotor plus the tail rotor. The user must input whether the flight condition is hover only, forward only, vertical only or forward and vertical. FLITE utilizes the Standard Data Set for storage registers 00 through 45, and the non-standard registers listed in Additional Storage Registers.

To execute the program and view only the power results, enter XEQ FLITE. To execute the program and view power and parameter results, enter XEQ FLITE+. This program is also written so that each parameter may be calculated individually, as long as the user executes DATA first (XEQ DATA), or the data is already in the proper storage registers. To execute the individual subroutine, follow the instructions listed in Subroutines to Determine Individual Parameters.

Additional Programs Required: None

Equations

$A_D = \pi R^2$ Ref 1, Eqn 2-36

$V_T = \Omega R$ Ref 1, Eqn 2-22

$\mu = V_f/V_T$ Ref 1, Page 123

$C_T = W/(\rho A V_T^2)$ Ref 1, Eqn 2-23

$B = 1 - (2C_T)^{.5}/b$ Ref 1, Eqn 3-1

$\sigma = bc/\pi R$ Ref 1, Eqn 2-37

$c_e = c_1 + .25(c_o - c_1)(1 - a^4)/)1 - a)$ Ref 1, Page 55

$H = 1 - H_p \cdot 6.875 \cdot 10^{-6}))^{5.2561}/((T^{o}F - 32) \cdot .555 + 273.16)$ Ref 2, Page 55

$H_\rho = 1 - (H \cdot 288.16)^{.23496}/6.875 \cdot 10^{-6}$ Ref 2, Page 55

FLITE

$\rho = 0.0023769 \cdot (1 - 6.875 \cdot 10^{-6} H_\rho)^{4.2561}$ Ref 2, Page 55

$P_{i_{IGE}}/P_{i_{OGE}} = -.1276)h/D)^4 + .7070(h/D)^3 - 1.4569(h/D)^2 + 1.3434(h/D) + 0.5147$ Ref 1, Eqn 3-8

$v_{i_h} = (W/2\rho A_D)^{.5}$ Ref 1, Eqn 4-1

$v_{i_v} = .5(-V_v + \{V_v + (4v_{i_h})^{.5}\})^{.5}$ Ref 1, Eqn 4-13

$v_{i_T} = -V_f^2/2 + ((V_f^2/2)^2 + v_{i_h}^4)^{.5})^5$ Ref 1, Eqn 4-34

$v_i = v_{i_T}$ or v_{i_v} or $v_{i_{Tv}}$ depending on flight conition

$P_i = (Wv_i)/(550)$ Ref 1, Eqn 4-2

$P_{i_{TL}} = (Wv_i)/(550 \cdot B)$ Ref 1, Eqn 3-6

$P_{i_{(TL+GE)}} = P_{i_{TL}} \cdot (P_{i_{IGE}}/P_{i_{OGE}})$ Ref 1, Eqn 4-3

$P_o = \sigma C_{d_o} \rho A_D V_T^3 (1 + 4.3\mu^2)/4400$ Ref 1, 4-53

$P_p = .5\rho((V_f^3 F_f) + (V_v^3 F_v))/550$ Ref 1, Eqn 4-62

$P_c = (WV_v)/550$ Ref 1, Eqn 4-62

$P_T = P_{i_{(TL+GE)}} + P_o + P_p + P_c$ Ref 1, Eqn 4-64

$C_{T_{(TR)}} = P_{T_{MR}}/\rho A_{D_{TR}} \ell\Omega_{TR}$ Ref 1, Page 145

$B_{(TR)} = 1 - (2CT_{TR})^{.5}/b_{TR}$ Ref 1, Page 146

$v_{i_{h(TR)}} = (P_{T_{MR}}/2\rho A_{D_{TR}} \ell\Omega_{MR}))^{.5}$ Ref 1, Eqn 5-8

$v_{i_{T(TR)}} = ((-V_f^2/2 + ((V_f^2/2)^2 + v_{i_{h(TR)}}^2))^{.5})^{.5}$ Ref 1, Eqn 5-9

$P_{i_{(TR)}} = P_{T_{MR}} v_{i_{T(TR)}}/(\ell\Omega_{MR} \cdot 550)$ Ref 1, Eqn 5-11

$P_{i_{(TL+GE)}} = P_{i_{TR}}/(B_{TR} \cdot 550)$ Ref 1, Eqn 5-4

$$P_{o_{(TR)}} = \sigma C_{d_o} \rho A_D V_T^3 (1 + 4.3\mu^2)/4400 \quad _{TR}$$ Ref 1, Eqn 5-12

$$P_{T_{(TR)}} = P_{i_{TR}} + P_{o_{TR}}$$ Ref 1, Eqn 5-14

$$P_{T_{(AC)}} = P_{T_{MR}} + P_{T_{TR}}$$

Additional Storage Registers

Storage Register	Stored Quantity
46	SD - Solidity of Main Rotor
47	SD(TR) - Solidity of Tail Rotor
48	AR - Advance Ratio of Main Rotor
49	AR(TR) Advance Ratio of Tail Rotor
50	PI - Induced Power
51	PT(TL) - Induced power with Tip Loss
52	PI(TR) - Tail Rotor Induced Power

Subroutines to Determine Individual Parameters:

***** To AVIEW the answers for these subroutines, set flag 07 (SF 07) ****

Input data -- XEQ DATA

Density -- XEQ DATA : XEQ PA

Disk Area (MR) -- XEQ DATA : XEQ AD

Disk Area (TR) -- XEQ DATA : XEQ ADTR

Tip Velocity -- XEQ DATA : XEQ VT

Tip Velocity (TR) -- XEQ DATA : XEQ VTTR

Advance Ratio (MR) -- XEQ DATA : Store VF (ft/sec) in Reg 18 : XEQ AR

Advance Ratio (TR) -- XEQ DATA : Store VF (ft/sec) in Reg 18 : XEQ ARTR

Solidity (MR) -- XEQ DATA : XEQ SD

Solidity (TR) -- XEQ DATA : XEQ SDTR

Coefficient of Thrust (MR -- XEQ DATA : XEQ CT

Tip Loss (MR) -- XEQ DATA : XEQ CT : XEX TL

Ground Effect -- XEQ DATA : XEQ GE

**** The following parameters can only be determined by executing FLITE and then recalling the storage register for that parameter ****

Coefficient of Thrust (TR) -- XEQ FLITE : RCL 30

Tip Loss (TR) -- XEQ FLITE : RCL 34

Induced Velocity (MR) -- XEQ FLITE : RCL 31 (value will be for chosen flight condition)

Induced Velocity (TR) -- XEQ FLITE : RCL 32 (value will be for chosen flight condition)

				Size 060
	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Initialize program		XEQ FLITE	MR ONLY?
2.	You may determine power requirements for main rotor only or main and tail rotor MR ONLY (1) MR + TR (0)	0	R/S	MR AND TR (pause) NEED DATA?
3.	Do you need to input any data? Yes (1) No (0) If No, go to Step 21a	1	R/S	REC?
4.	Is main rotor blade rectangular?	1	R/S	c0=?
5	Yes - Input value of chord (ft). Go to Step 8	1.75	R/S	R=?
5a	No - Input root chord (ft)	-	-	(Cl=?)
6.	No - Input tip chord (ft)	-	-	(a=?)
7.	Input fractional distance where taper begins	-	-	ce=
8.	Continue from Step 5		R/S	R=?
9.	Input MR Radius (ft)	26.8	R/S	RV=?
10.	Input MR Rotational velocity (rad/sec)	27.0	R/S	Cdo=?
11.	Input MR coefficient of drag	.008	R/S	b=?

INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
12. Input number of Main Rotor blades	4	R/S	R(TR)=?
13. Input TR radius (ft)	5.5	R/S	c(TR)=?
14. Input TR chord (ft)	.81	R/S	RV(TR)=?
15. Input TR rotational velocity (rad/sec)	124.6	R/S	Cdo(TR)=?
16. Input TR Coefficient of Drag	.008	R/S	b(TR)=?
17. Input number TR blades	4	R/S	L=?
18. Input tail length (ft)	31.5	R/S	W=?
18a. Continue from Step 12			W=?
19. Input gross weight (lbs)	20000	R/S	RTR HT=?
20. Input height of MR above skid/wheel (ft)	11.2	R/S	SKID HT=?
21. Input height of skids above ground (ft)	2500	R/S	HOVER ONLY?
21a. Continue from Step 3			HOVER ONLY?
22. Is problem for hover flight only? Yes (1) No (0) If Yes go to Step 28a	0	R/S	FWD ONLY?
23. Is problem for forward flight only? Yes (1) No (0) If Yes go to Step 25	0	R/S	VERT ONLY? VF=? (Yes)
24. Is problem for vertical flight only? Yes (1) No (0) If Yes go to Step 27	0	R/S	FWD + VERT (pause) VF=? VV=? (Yes)

	INSTRUCTION	INPUT	FUNCTION	DISPLAY
25.	Input Forward velocity (kts) (Must input each time through program	50	R/S	FF=?
26.	Input forward Equiv. Flat Plate Area (ft^2) (Must input each time) If FWD ONLY go to Step 28a.	25.7	R/S	VV=?
27.	Input vertical velocity (fpm). (Must input each time)	200	R/S	FV=?
28.	Input vert. Equiv. Flat Area (ft^2). (Must input each time)	30.8	R/S	PA?
28a.	Continue			PA?
29.	Are you using Pressure Altitude? Yes (1) No (0) If Yes go to Step 30a.	0	R/S	DA=?
29.	If you are running FLITE+, you will get output for DEN, AD, VT AR, CT, B, SD and VI			DEN = * * VI=
30.	Input density alt. (ft) Go to Step 31	2500	R/S	PI=773.6
30a.	Input Temp (^oF)		R/S	PI=
31.	Output induced power w/o TL or GE (SHP)		R/S	PI(TL)=798
31a.	If you are running FLITE+, you will get output for GE.			GE=
32.	Ouput induced power w/TL (SHP)		R/S	PI(TL+GE)= 798.3
33.	Output induced power w/(TL+GE) SHP)		R/S	PO=302.0

	INSTRUCTION	INPUT	FUNCTION	DISPLAY
34.	Output profile power (SHP)		R/S	PP=33.1
35.	Output parasite power (SHP)		R/S	PC= 121.2
36.	Output climb power (SHP)		R/S	PT(MR)= 1252.6
37.	Output total power (SHP)			PI(TR+TL)= 33.6
37a.	If you are running FLITE+, you get output for ADTR, VTTR, ARTR SDTR, CTTR, BTR, VITR and PI(TR)			ADTR= * * * PI(TR)=
37b.	If MR only, go to Step 42.		R/S	CHANGE?
38.	Output induced power (TR) w/TL (SHP)		R/S	PO(TR)=24.5
39.	Output profile power (TR) (SHP)		R/S	PT(TR)=58.2
40.	Output total power (TR) (SHP)		R/S	PT(AC)= 1310.7
41.	Output total power for AC MR + TR (SHP)		R/S	CHANGE?
42.	Do you wish to change any Input Data? Yes (1) No (1) Yes will restart No will stop program See NOTE	0	R/S	0.0000

Note: To make changes there is no need to input all the data again. Just change those values which require changing. When other values are requested, just press R/S.

*** THIS DOES NOT APPLY TO VF, VV, FF or FV ***

FLITE

```
01*LBL "FLITE"
02 CF 07
03 GTO 10
04*LBL "FLITE+"
05 SF 07
06*LBL 10
07 CF 01
08 CF 02
09 CF 03
10 CF 04
11 CF 05
12 CF 06
13 0
14 STO 12
15 STO 13
16 STO 18
17 STO 19
18 FIX 3
19 "MF ONLY ?"
20 PROMPT
21 X>0?
22 GTO 01
23 "*MF*TP*"
24 AVIEW
25 PSE
26 GTO 02
27*LBL 01
28 SF 01
29*LBL 02
30 XEQ "DATA"
31 "HOVER ONLY ?"
32 PROMPT
33 X>0?
34 GTO 07
35 SF 04
36 "FWD ONLY ?"
37 PROMPT
38 X>0?
39 GTO 08
40 SF 05
41 "VERT ONLY ?"
42 PROMPT
43 X>0?
44 GTO 09
45 "*FWD*VERT*"
46 AVIEW
47 PSE
48 SF 06
49 CF 04
50 CF 05
51*LBL 08
52 "VF(KTS)=?"
53 PROMPT
54 1.68889
55 *
56 STO 18
57 "FF=?"
58 PROMPT
59 STO 12
60 FS? 04
61 GTO 07
62 GTO 09
63*LBL 09
64 "VV(FPM)=?"
65 PROMPT
66 60
67 /
68 STO 19
69 "FV=?"
70 PROMPT
71 STO 13
72*LBL 07
73 XEQ "PA"
74 XEQ "AD"
75 XEQ "VT"
76 XEQ "AP"
77 XEQ "CT"
78 XEQ "TL"
79 XEQ "SD"
80 XEQ "VIH"
81 FS? 05
82 GTO 13
83 XEQ "VIT"
84 FS? 06
85 GTO 14
86 GTO 15
87*LBL 13
88 XEQ "VIV"
89 GTO 15
90*LBL 14
91 XEQ "VITV"
92*LBL 15
93 "VI="
94 FS? 07
95 XEQ "S"
96 FIX 1
97 XEQ "PI"
98 XEQ "PO"
99 XEQ "PP"
100 XEQ "PC"
101 XEQ "PT"
102 FIX 3
103 FS? 01
104 GTO 20
105 XEQ "TR"
106*LBL 20
107 "CHANGE ?"
108 PROMPT
109 X>0?
110 GTO 11
111 GTO 12
112*LBL 11
113 FS? 07
114 XEQ "FLITE+"
115 XEQ "FLITE"
116*LBL 12
117 STOP
118*LBL "DATA"
119 "NEED DATA ?"
120 PROMPT
121 X=0?
122 GTO 03
123 "REC ?"
124 PROMPT
125 X>0?
126 GTO 04
127 XEQ "ECHORD"
128 GTO 05
129*LBL 04
130 RCL 01
131 "c=?"
132 PROMPT
133*LBL 05
134 STO 01
135 RCL 00
136 "P=?"
137 PROMPT
138 STO 09
139 RCL 02
140 "PV=?"
141 PROMPT
142 STO 02
143 RCL 03
144 "Cd0=?"
145 PROMPT
146 STO 03
147 RCL 04
148 "b=?"
149 PROMPT
150 STO 04
151 FS? 01
152 GTO 06
153 RCL 05
154 "P(TR)=?"
155 PROMPT
156 STO 05
157 RCL 06
158 "c(TR)=?"
159 PROMPT
160 STO 06
161 RCL 07
162 "RV(TR)=?"
163 PROMPT
164 STO 07
165 RCL 08
166 "Cd0(TR)=?"
167 PROMPT
168 STO 08
169 RCL 09
170 "b(TR)=?"
171 PROMPT
172 STO 09
173 RCL 10
174 "L=?"
175 PROMPT
176 STO 10
177*LBL 06
178 RCL 11
179 "W=?"
180 PROMPT
181 STO 11
182 RCL 14
183 "PTP HT=?"
184 PROMPT
185 STO 14
186 RCL 25
187 "SKID HT=?"
188 PROMPT
189 STO 25
190*LBL 03
191 RTN
192*LBL "ECHORD"
193 RCL 15
194 "c0=?"
195 PROMPT
196 STO 15
197 RCL 16
198 "cL=?"
199 PROMPT
200 STO 16
```

FLITE

```
001 RCL 17
002 "a=?"
003 PROMPT
004 STO 17
005 4
006 Y↑X
007 CHS
008 1
009 +
010 RCL 15
011 RCL 16
012 -
013 *
014 RCL 17
015 CHS
016 1
017 +
018 /
019 4
020 /
021 RCL 16
022 +
023 STO 01
024 "ce="
025 XEQ "S"
026 RTN
027*LBL "PA"
028 "PA ?"
029 PROMPT
030 X=0?
031 GTO "DA"
032 "PA=?"
033 PROMPT
034 STO 20
035 6.875 E-06
036 *
037 CHS
038 1
039 +
040 5.2561
041 Y↑X
042 "TEMP(F)=?"
043 PROMPT
044 STO 21
045 32
046 -
047 .5555
048 *
049 273.16
050 +
051 STO 21
052 /
053 288.16
054 *
055 .23496
056 Y↑X
057 CHS
058 1
059 +
060 6.875 E-06
061 /
062 STO 20
063 GTO "DEN"
064*LBL "DA"
065 "DA=?"
066 PROMPT
067 STO 20
068*LBL "DEN"
069 RCL 20
070 6.875 E-06
071 *
072 CHS
073 1
074 +
075 ENTER↑
076 4.2561
077 Y↑X
078 .0023769
079 *
080 STO 22
081 FIX 6
082 "DEN="
083 FS? 07
084 XEQ "S"
085 FIX 3
086 RTN
087*LBL "AD"
088 RCL 00
089 X↑2
090 PI
091 *
092 STO 23
093 "AD="
094 FS? 07
095 XEQ "S"
096 RTN
097*LBL "VT"
098 RCL 02
099 RCL 00
100 *
101 STO 27
102 "VT="
103 FS? 07
104 XEQ "S"
105 RTN
106*LBL "AR"
107 RCL 18
108 RCL 27
109 /
110 STO 48
111 "AR="
112 FS? 07
113 XEQ "S"
114 RTN
115*LBL "CT"
116 RCL 22
117 PI
118 *
119 RCL 00
120 X↑2
121 *
122 RCL 09
123 RCL 02
124 *
125 X↑2
126 *
127 1/X
128 RCL 11
129 *
130 STO 29
131 FIX 6
132 "CT="
133 FS? 07
134 XEQ "S"
135 FIX 3
136 RTN
137*LBL "TL"
138 FIX 3
139 RCL 29
140 2
141 *
142 SQRT
143 RCL 04
144 /
145 CHS
146 1
147 +
148 STO 33
149 "θ="
150 FS? 07
151 XEQ "S"
152 RTN
153*LBL "SD"
154 RCL 04
155 RCL 01
156 *
157 RCL 00
158 /
159 PI
160 /
161 STO 46
162 "SD="
163 FS? 07
164 XEQ "S"
165 RTN
166*LBL "GE"
167 RCL 25
168 RCL 14
169 +
170 RCL 00
171 2
172 *
173 /
174 STO 26
175 1.55
176 -
177 X>0?
178 GTO 18
179 RCL 26
180 1.3432
181 *
182 RCL 26
183 X↑2
184 -1.4569
185 *
186 +
187 RCL 26
188 3
189 Y↑X
190 .7000
191 *
192 +
193 RCL 26
194 4
195 Y↑X
196 -.1276
197 *
198 +
199 .514
200 +
```

```
001 GTO 19
002*LBL 18
003 1
004*LBL 19
005 STO 45
006 "GE="
007 FS? 07
008 XEQ "S"
009 RTN
010*LBL "VIH"
011 RCL 11
012 2
013 /
014 RCL 22
015 /
016 RCL 23
017 /
018 SQRT
019 STO 31
020 RTN
021*LBL "VIT"
022 RCL 18
023 X↑2
024 2
025 /
026 STO 53
027 X↑2
028 RCL 31
029 4
030 Y↑X
031 +
032 SQRT
033 RCL 53
034 -
035 SQRT
036 STO 31
037 RTN
038*LBL "VIV"
039 RCL 31
040 X↑2
041 4
042 *
043 RCL 19
044 X↑2
045 +
046 SQRT
047 RCL 19
048 -
049 2
050 /
051 STO 31
052 RTN
053*LBL "VITV"
054 RCL 19
055 2
056 /
057 CHS
058 RCL 31
059 +
060 STO 31
061 RTN
062*LBL "PI"
063 RCL 11
064 RCL 31
065 *
066 550
067 /
068 STO 50
069 "PI="
070 XEQ "S"
071 RCL 33
072 /
073 STO 51
074 "PI(TL)="
075 XEQ "S"
076 XEQ "GE"
077 RCL 51
078 *
079 STO 35
080 "PI(TL+GE)="
081 XEQ "S"
082 RTN
083*LBL "PD"
084 RCL 48
085 X↑2
086 4.3
087 *
088 1
089 +
090 RCL 27
091 3
092 Y↑X
093 *
094 RCL 23
095 *
096 RCL 22
097 *
098 RCL 03
099 *
100 RCL 46
101 *
102 4400
103 /
104 STO 36
105 "PD="
106 XEQ "S"
107 RTN
108*LBL "PP"
109 RCL 18
110 3
111 Y↑X
112 RCL 12
113 *
114 RCL 19
115 3
116 Y↑X
117 RCL 13
118 *
119 +
120 RCL 22
121 *
122 1100
123 /
124 STO 37
125 X=0?
126 GTO 16
127 "PP="
128 XEQ "S"
129*LBL 16
130 RTN
131*LBL "PC"
132 RCL 11
133 RCL 19
134 *
135 550
136 /
137 STO 38
138 X=0?
139 GTO 17
140 "PC="
141 XEQ "S"
142*LBL 17
143 RTN
144*LBL "PT"
145 RCL 35
146 RCL 36
147 +
148 RCL 37
149 +
150 RCL 38
151 +
152 STO 39
153 "PT(MR)="
154 XEQ "S"
155 RTN
156*LBL "TP"
157 XEQ "ADTR"
158 XEQ "VTTR"
159 XEQ "APTR"
160 XEQ "SDTR"
161 RCL 39
162 550
163 *
164 RCL 02
165 /
166 RCL 10
167 /
168 STO 40
169 RCL 24
170 /
171 RCL 22
172 /
173 STO 55
174 RCL 28
175 X↑2
176 /
177 STO 30
178 FIX 6
179 "CTTR="
180 FS? 07
181 XEQ "S"
182 FIX 3
183 2
184 *
185 SQRT
186 RCL 09
187 /
188 CHS
189 1
190 +
191 STO 34
192 "BTR="
193 FS? 07
194 XEQ "S"
195 RCL 55
196 2
197 /
198 X↑2
199 RCL 53
200 X↑2
```

```
601 +
602 SQRT
603 RCL 53
604 -
605 SQRT
606 STO 32
607 "VITP="
608 FS? 07
609 XEQ "S"
610 RCL 40
611 *
612 550
613 /
614 STO 52
615 FIX 1
616 "PI(TR)="
617 FS? 07
618 XEQ "S"
619 RCL 34
620 /
621 STO 41
622 "PI(TR+TL)="
623 XEQ "S"
624 XEQ "PDTR"
625 "PD(TR)="
626 XEQ "S"
627 +
628 STO 43
629 "PT(TR)="
630 XEQ "S"
631 RCL 39
632 +
633 STO 44
634 "PT(AC)="
635 XEQ "S"
636 FIX 3
637 RTN
638*LBL "ADTR"
639 RCL 05
640 X↑2
641 PI
642 *
643 STO 24
644 "ADTR="
645 FS? 07
646 XEQ "S"
647 RTN
648*LBL "VTTR"
649 RCL 07
650 RCL 05
651 *
652 STO 28
653 "VTTR="
654 FS? 07
655 XEQ "S"
656 RTN
657*LBL "ARTR"
658 RCL 18
659 RCL 28
660 /
661 STO 49
662 "ARTR="
663 FS? 07
664 XEQ "S"
665 RTN
666*LBL "SDTR"
667 RCL 09
668 RCL 06
669 *
670 RCL 05
671 /
672 PI
673 /
674 STO 47
675 "SDTR="
676 FS? 07
677 XEQ "S"
678 RTN
679*LBL "PDTR"
680 RCL 49
681 X↑2
682 4.3
683 *
684 1
685 +
686 RCL 28
687 3
688 Y↑X
689 *
690 RCL 24
691 *
692 RCL 22
693 *
694 RCL 09
695 *
696 RCL 47
697 *
698 4400
699 /
700 STO 42
701 RTN
702*LBL "S"
703 ARCL X
704 AVIEW
705 STOP
706 RTN
707 END
```

FUEL

Fuel Flow as a function of Power

Introduction: This program determines the relationship between fuel flow rate and power required - Phantom Shaft Horsepower (PSHP) and the fuel flow rate ($\dot{W}_f$) at specified velocities. This program runs in conjunction with POWER which must be loaded and the geometric and flight parameters loaded through step 19 in he step-by-step instructions. The input parameters are the number of engines (NENG), $\hat{\alpha}$ (the fuel flow rate versus SHP ordinate intercept) and $\hat{\beta}$ (the slope of the fuel flow rate versus SHP curve). The standard data set is used as well as those listed below.

Additional Programs Required: POWER

Equations

$$\dot{W}_f = (PSHP + P_T)\cdot\hat{\beta} \qquad \text{Ref 2, Eqn 7D-11}$$

$$PSHP = NENG\cdot\hat{\alpha}\cdot\delta\cdot\sqrt{\theta}/\hat{\beta} \qquad \text{Ref 2, Eqn 7D-9}$$

Additional Storage Registers

Storage Registers	Stored Quantity
49	PSHP - Phantom Shaft Horsepower
55	NENG - Number of engines
56	$\hat{\alpha}$ - Fuel Flow Rate vs SHP intercept
57	$\hat{\beta}$ - Slope of Fuel Flow Rate vs SHP curve

FUEL

			SIZE 060
INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1. Initialize the program		XEQ FUEL	*LOAD POWER*
2. Reminder flashes			NENG=?
3. Enter the number of engines	1	R/S	ALPHA HAT=?
4. Enter the sea level intercept of the fuel flow rate vs SHP	88.5	R/S	BETA HAT=?
5. Enter the slope of the fuel flow rate versus SHP curve	0.5	R/S	PSHP=177.0
6. Display Phantom horsepower		R/S	VF=?
7. Enter the forward velocity (kts)	100	R/S	VV=?
8. Enter the vertical velocity (fpm)	0	R/S	W DOT=652.6
9. Display Fuel Flow Rate (lb/hr)		R/S	ANOTHER VF?
10. Exit the program	0	R/S	0.0

FUEL

```
01*LBL "FUEL"
02 "*LOAD POWER*"
03 AVIEW
04 PSE
05 "NENG=?"
06 PROMPT
07 STO 55
08 "ALPHA HAT=?"
09 PROMPT
10 STO 56
11 "BETA HAT=?"
12 PROMPT
13 STO 57
14 1/X
15 *
16 *
17 RCL 47
18 *
19 RCL 21
20 518.688
21 /
22 SQRT
23 *
24 STO 49
25 "PSHP="
26 ARCL X
27 AVIEW
28 STOP
29*LBL 10
30 SF 03
31 CF 02
32 XEQ "VF"
33 RCL 49
34 RCL 44
35 +
36 RCL 57
37 *
38 "W DOT="
39 ARCL X
40 AVIEW
41 STOP
42 "ANOTHER VF?"
43 PROMPT
44 X=0?
45 GTO "COMP"
46 GTO 10
47*LBL "COMP"
48 CF 03
49 CF 02
50 END
```

HSE

High Speed Effects

Introduction: This program calculates the high speed effects including both retreating blade stall and advancing blade Mach effects. Using main rotor geometric design parameters and the forward velocity as input variables, the collective and cyclic angles and the angle of attack at the tip of the blade at the 90 degree and 270 degree azimuth are determined. The change of power due to stall effects (*P<S>) and the change in power due to Mach effects (*P<M>) are computed and added to the total power required to furnish a corrected power required (P<C>). The collectve, cyclic and blade angles of attack are not displayed, but may be obtained by recalling the applicable data register.

Due to the difficulties with the Retreating Blade Stall equations as listed in Ref. 1, this program assumes that the Coefficient of Profile Drag is doubled when the retreating blade tip angle of attack exceed the maximum angle by 4°, and also assumes that the change in power due to stall is linear.

This program must be run following POWER and uses the same standard data registers as POWER. Additional data registers are as shown below.

Maximum angle of attack (blade stall angle) must be entered, and a value other than zero must be entered for the blade twist. The normal range of values for twist is from -7 to -18 degrees.

Additional Programs Required: POWER

Equations:

$$2C_T/a\sigma = \lambda T_1 + \theta_0 T_2 + \theta_T T_3 + \theta_2 T_4 \qquad \text{Ref 1, Eqn 8-19}$$

$$0 = \lambda A_{11} + \theta_0 A_{12} + \theta_T A_{13} + \theta_2 A_{14} \qquad \text{Ref 1, Eqn 8-20}$$

$$\lambda = (V\alpha_3 - v_w)/V_T$$

$$\alpha_3 = \tan^{-1}(D_p/W) \qquad \text{Ref 1, Eqn 8-4}$$

$$\mu = V_f/V_T \qquad \text{Ref 1, Eqn 8-2}$$

Equations (Continued)

$T_1 = .5(B^2 + .5\mu^2)$ Ref 1, Eqn 8-17

$T_2 = (.33B^3 + .5\mu^2 B)$ Ref 1, Eqn 8-17

$T_3 = .25B^2(B^2 + \mu^2)$ Ref 1, Eqn 8-17

$T_4 = .5\mu(B^2 + .25\mu^2)$ Ref 1, Eqn 8-17

$A_{11} = 4(\mu B^2/2 - \mu^3/8)/B^2(B^2 - .5\mu^2)$ Ref 1, Eqn 8-18

$A_{12} = 8\mu B/3(B^2 - .5\mu^2)$ Ref 1, Eqn 8-18

$A_{13} = 2\mu B^2/(B^2 - .5\mu^2)$ Ref 1, Eqn 8-18

$A_{14} = (B^2 + 1.5\mu^2)/(B^2 - .5\mu^2)$ Ref 1, Eqn 8-18

θ_0 = Collective Pitch Angle

θ_2 = Longitudinal Cyclic Pitch Angle

θ_T = Blade Twist Angle

$C_{do_s} = 2\, C_{do}\, (\alpha^0_{90} - \alpha^0_{max})/4$

$*P<S> = P_o[C_{do_s}/C_{do}]$

$M_{crit} = 0.71 - |2.3|\alpha_{90}$ (rad) Ref 1, Eqn 8-30

$M_{90} = M_{tip}(1 + \mu)$ Ref 1, Eqn 8-29

$M_d = M_{90} - M_{crit} - 0.06$ Ref 1, Eqn 8-21

$C_{p_m} = (0.012\Delta M_d + 0.1\Delta M_d^3)$ Ref 1, Eqn 8-31

$*P<M> = C_{p_m} A\rho V_T^3$ Ref 1, Eqn 3-25

Additional Storage Registers

Storage Register	Stored Quantity
15	σ - Solidity
16	~~*P<M> - Change in Power due to Mach effects~~
17	ΔM_d - Excess Mach over corrected critical value
46	θ_T - Twist angle
47	λ = Inflow ratio
48	μ = Advance ratio (V_f/V_T)

The Following Registers are used for storage after line 160

49	θ_0 (Collective Pitch)
50	α_{270} (Retreating Blade Tip Angle of Attack)
52	*P<S> - Change in Power due to Blade Stall Effects
53	*P<M> - Change in Power due to Compressibility
54	α_{90} (Advancing Blade Tip Angle of Attack)
55	θ_2 (Longitudinal Cyclic Pitch)

HSE

SIZE 060

INSTRUCTION	INPUT	FUNCTION	DISPLAY
1. Initialize Program		XEQ HSE	TWIST<DEG>=?
2. Input Twist Angle (Degrees)	-9	R/S	a<MAX>=?x
3. Input max angle of attack (Degrees)	12.5	R/S	BLADE STALL
4. Output change in power due to stall (SHP)		R/S	*P<S>=668.2
5. Output change in power due to Mach (SHP)		R/S	*P<M>=393.8
6. Output total power including high speed effects (SHP)		R/S	PT<C>=3,3157.2

Note: This program may be accessed directly from POWER by answering the question "HI SPD?" with a Yes (Input <1>), or it may be run directly as shown above, provided that either POWER has previously be run or the data registers 1 through 19 plus 44 have been loaded with the correct values.

HSE

```
01*LBL "HSE"
02 RCL 46
03 57.3
04 *
05 "TWIST<DEG>=?"
06 PROMPT
07 57.3
08 /
09 STO 46
10 RCL 16
11 "a<MAX>=?"
12 PROMPT
13 STO 16
14 RCL 37
15 RCL 11
16 /
17 550
18 *
19 RCL 31
20 2
21 *
22 +
23 RCL 27
24 /
25 CHS
26 STO 47
27 RCL 18
28 RCL 27
29 /
30 STO 48
31 X↑2
32 STO 55
33 RCL 33
34 X↑2
35 STO 54
36 RCL 55
37 2
38 /
39 CHS
40 RCL 54
41 +
42 STO 49
43 RCL 48
44 RCL 55
45 *
46 8
47 /
48 CHS
49 RCL 54
50 RCL 48
51 *
52 2
53 /
54 +
55 4
56 *
57 RCL 49
58 /
59 RCL 54
60 /
61 STO 50
62 RCL 33
63 RCL 48
64 *
65 2.6667
66 *
67 RCL 49
68 /
69 STO 51
70 RCL 48
71 RCL 54
72 *
73 2
74 *
75 RCL 49
76 /
77 STO 52
78 RCL 55
79 1.5
80 *
81 RCL 54
82 +
83 RCL 49
84 /
85 STO 53
86 RCL 55
87 2
88 /
89 RCL 54
90 +
91 2
92 /
93 STO 56
94 RCL 55
95 2
96 /
97 RCL 54
98 3
99 /
100 +
101 RCL 33
102 *
103 STO 57
104 RCL 54
105 RCL 55
106 +
107 RCL 54
108 *
109 4
110 /
111 STO 58
112 RCL 55
113 4
114 /
115 RCL 54
116 +
117 RCL 48
118 *
119 2
120 /
121 STO 59
122 RCL 29
123 .15
124 *
125 RCL 15
126 /
127 STO 55
128 RCL 47
129 RCL 56
130 *
131 ST- 55
132 RCL 46
133 RCL 58
134 *
135 ST- 55
136 RCL 55
137 RCL 51
138 *
139 STO 17
140 RCL 47
141 RCL 50
142 *
143 RCL 46
144 RCL 52
145 *
146 RCL 57
147 *
148 RCL 17
149 +
150 CHS
```

```
151 STO 17
152 RCL 53
153 RCL 57
154 *
155 RCL 59
156 RCL 51
157 *
158 -
159 1/X
160 RCL 17
161 *
162 STO 55
163 RCL 53
164 *
165 RCL 47
166 RCL 50
167 *
168 +
169 RCL 46
170 RCL 52
171 *
172 +
173 RCL 51
174 /
175 CHS
176 STO 49
177 RCL 55
178 -
179 RCL 46
180 +
181 RCL 48
182 1
183 +
184 1/X
185 RCL 47
186 *
187 +
188 57.3
189 *
190 STO 50
191 RCL 16
192 -
193 0
194 X>Y?
195 GTO "NS"
196 X<>Y
197 "BLADE STALL="
198 AVIEW
199 PSE
200 4
201 /
202 RCL 48
203 X↑2
204 4.3
205 *
206 1
207 +
208 1/X
209 RCL 36
210 *
211 *
212 STO 53
213 "P(S)="
214 ARCL X
215 AVIEW
216 STOP
217*LBL "MP"
218 RCL 22
219 .11748
220 Y↑X
221 2257.3
222 *
223 1/X
224 RCL 18
225 RCL 27
226 +
227 *
228 .77
229 -
230 RCL 54
231 .0401
232 *
233 +
234 STO 56
235 0
236 X>Y?
237 GTO "NM"
238 X<>Y
239 3
240 Y↑X
241 .1
242 *
243 RCL 56
244 .012
245 *
246 +
247 RCL 15
248 *
249 RCL 22
250 *
251 RCL 27
252 *
253 RCL 27
254 3
255 Y↑X
256 *
257 550
259 /
259 STO 52
260 GTO "MH"
261*LBL "NM"
262 0
263 STO 52
264*LBL "MH"
265 RCL 52
266 "P(M)="
267 ARCL X
268 AVIEW
269 STOP
270 RCL 52
271 RCL 53
272 +
273 RCL 44
274 +
275 "P(HSE)="
276 ARCL X
277 AVIEW
278 STOP
279 GTO "MF"
280*LBL "NS"
281 "NO STALL"
282 AVIEW
283 STOP
284 GTO "MP"
285 .END.
```

POWER

Helicopter Power Requirements

Introduction: The user inputs the basic geometric and flight parameters of the helicopter and the program determines the steady state power requirements to maintain that condition. High speed effects are not included but may be determined by running the high speed effects program (HSE) following this program. POWER requires no other subroutines to operate. It utilizes the Standard Data Set for storage registers 00 through 45, and additionally those listed below.

Additional Programs Required: None

Equations:

$\{(h/d) - 1.55\} < 0$? (In Ground Effect?) Ref 1, Fig 2, pg 67

$$(P_i/P_{i_{OGE}}) = 0.5147 + 1.3432(h/D) - 1.4569(h/D)^2 + 0.7080(h/D)^3 - 0.1276(h/D)^4$$ Ref 1, Eqn 3-8

$$P_{alt} = P_{ssl}(1 - 6.875x10^{-6} \cdot H)^{5.2561}$$ Ref 2, Eqn 7D-5

$$_{alt} = _{ssl}(1 - 6.8755x10^{-6} \cdot H)^{4.2561}$$ Ref 2, Eqn 7D-6

$$C_T = W/\rho A(\pi R)^2$$ Ref 1, Eqn 2-23

$$B = 1 - \sqrt{2C_T}/b$$ Ref 1, Eqn 3-1

$$v_i = (W/2\rho A)^{.5}$$ Ref 1, Eqn 2-15

$$P_i = (P_i/P_{i_{OGE}})(Tv_i/(550 \cdot B)\left\{\{1+.25(V_f^2/v_i^2)\} - .5(v_f^2/v_i^2)\right\}^{.5}$$ Ref 1, Eqn 4-41

$$P_o = (C_{d_o} \rho b\ c\ R\ V_T^3/4400)(1 + 4.3\mu^2)$$ Ref 1, Eqn 4-53

$$P_p = V_f^3 F_f \rho/1100$$ Ref 1, Eqn 4-56

$$P_c = TV_v/550 + (.5\rho F_v V_v^3)/550$$ Ref 1, Eqn 4-28, Eqn 4-29

$$T_{TR} = 556\ P_{MR}/(\Omega_{MR} \ell_{TR})$$ Ref 1, Eqn 5-2

POWER

	INSTRUCTION	INPUT	FUNCTION	SIZE 060 DISPLAY
1.	Initialize Program		XEQ POWER	
2.	Reminder flashes		R/S	NEED DATA?
3.	Answer 1 for yes, 0 for no	1	R/S	W=?
4.	Input weight (lbs)	20000	R/S	RV=?
5.	Input main rotor rotational velocity (rad/sec)	27	R/S	b=?
6.	Input number of main rotor blades	4	R/S	c=?
7.	Input main rotor chord (ft)	1.75	R/S	Cdo=?
8.	Input main rotor drag coefficient	.008	R/S	R=?
9.	Input main rotor radius (ft)	26.8	R/S	FF=?
10.	Input forward flat plate area (ft^2)	25.7	R/S	FV=?
11.	Input vertical flat plate area (ft^2)	30.8	R/S	RV(TR)=?
12.	Input tail rotor rotational velocity (rad/sec)	124.6	R/S	b(TR)=?
13.	Input number of tail rotor blades	4	R/S	c(TR)=?
14.	Input tail rotor chord (ft)	.81	R/S	Cdo(TR)=?
15.	Input tail rotor drag coefficient	.008	R/S	R(TR)=?
16.	Input tail rotor radius (ft)	5.5	R/S	L(Tail)=?

	INSTRUCTION	INPUT	FUNCTION	DISPLAY
17.	Input length of tail (ft)	30.5	R/S	RTR HT=?
18.	Input rotor height above skid (ft)	11.2	R/S	SKID HT=?
19.	Input skid height above ground (ft)	100	R/S	PA?
20.	Do you know pressure altitude?			
	a. Answer 1 for yes	1	R/S	PA=?
	Input pressure altitude (ft)	0	R/S	TEMP(C)=?
	Input temperature	59	R/S	VF=?
or	b. Answer 0 for no	0	R/S	DA=?
	Input density altitude (ft)	0	R/S	VF=?
21.	Input forward 150 velocity (kt)		R/S	VV=?
22.	Input vertical 0 velocity (ft/min)		R/S	
23.	Output total aircraft power (SHP)		R/S	PT(AC)=1687.2
24.	Output main rotor power (SHP)			PT(MR)=1647.8
25.	Do you want high effects?			HI SPD?
26.	1 for Yes 0 for No		R/S	

POWER

```
01♦LBL "POWER"
02 FIX 1
03 CF 02
04 CF 03
05 "NEED DATA?"
06 PROMPT
07 X=0?
08 GTO "PGM"
09 RCL 11
10 "W=?"
11 PROMPT
12 STO 11
13 RCL 02
14 "RV=?"
15 PROMPT
16 STO 02
17 RCL 04
18 "b=?"
19 PROMPT
20 STO 04
21 RCL 01
22 "c=?"
23 PROMPT
24 STO 01
25 RCL 03
26 "Cd0=?"
27 PROMPT
28 STO 03
29 RCL 00
30 "R=?"
31 PROMPT
32 STO 00
33 RCL 12
34 "FF=?"
35 PROMPT
36 STO 12
37 RCL 13
38 "FV=?"
39 PROMPT
40 STO 13
41 RCL 07
42 "RV(TR)=?"
43 PROMPT
44 STO 07
45 RCL 09
46 "b(TR)=?"
47 PROMPT
48 STO 09
49 RCL 06
50 "c(TR)"
51 PROMPT
52 STO 06
53 RCL 08
54 "Cd0(TR)=?"
55 PROMPT
56 STO 08
57 RCL 05
58 "R(TR)=?"
59 PROMPT
60 STO 05
61 RCL 10
62 "L(TAIL)=?"
63 PROMPT
64 STO 10
65 RCL 14
66 "RTR HT=?"
67 PROMPT
68 STO 14
69♦LBL "PGM"
70 RCL 25
71 "SKID HT=?"
72 PROMPT
73 STO 25
74 RCL 14
75 +
76 RCL 00
77 /
78 2
79 /
80 STO 26
81 1.55
82 -
83 X<0?
84 GTO "GE"
85 1
86 STO 45
87 GTO "AREA"
88♦LBL "GE"
89 RCL 26
90 1.3432
91 *
92 RCL 26
93 X↑2
94 -1 4569
95 *
96 +
97 RCL 26
98 3
99 Y↑X
100 .7088
101 *
102 +
103 RCL 26
104 4
105 Y↑X
106 -.1276
107 *
108 +
109 .5147
110 +
111 STO 45
112♦LBL "AREA"
113 RCL 00
114 X↑2
115 PI
116 *
117 STO 23
118 RCL 05
119 X↑2
120 PI
121 *
122 STO 24
123♦LBL "VT"
124 RCL 00
125 RCL 02
126 *
127 STO 27
128 RCL 05
129 RCL 07
130 *
131 STO 28
132♦LBL "DA"
133 "PA?"
134 PROMPT
135 X=0?
136 GTO "DNA"
137 SF 05
138 "PA=?"
139 PROMPT
140 STO 20
141♦LBL "ICAO"
142 6.875 E-06
143 *
144 CHS
145 1
146 +
147 FS? 05
148 5.2561
149 FS? 04
150 4.2561
151 Y↑X
152 FS? 04
153 GTO "PHO"
154 FS? 05
155 STO 47
156♦LBL "TEMP"
157 "TEMP(F)=?"
158 PROMPT
159 459.688
160 +
161 STO 21
162 518.688
163 /
164 1/X
165 RCL 47
166 *
167 GTO "PHO"
168♦LBL "DNA"
169 "DA=?"
170 PROMPT
171 STO 20
172♦LBL "DEN"
173 SF 04
174 GTO "ICAO"
175♦LBL "PHO"
176 STO 51
177 .0023769
178 *
179 STO 22
180 FS? 05
181 GTO "VF"
182 RCL 20
183 6.875 E-06
184 *
185 CHS
186 1
187 +
188 518.688
189 *
190 STO 21
191 518.688
192 /
193 RCL 51
194 *
195 STO 47
196♦LBL "VF"
197 CF 05
198♦LBL "VV"
199 "VF=?"
200 PROMPT
```

POWER

```
201 1.68894
202 *
203 STO 18
204 "VV=?"
205 PROMPT
206 60
207 /
208 STO 19
209 CF 05
210 CF 04
211♦LBL "CT"
212 FS? 02
213 GTO 07
214 RCL 11
215 RCL 23
216 /
217 RCL 22
218 /
219 RCL 27
220 X↑2
221 /
222 STO 29
223 GTO "TL"
224♦LBL 07
225 RCL 40
226 RCL 24
227 /
228 RCL 22
229 /
230 RCL 28
231 X↑2
232 /
233 STO 30
234♦LBL "TL"
235 FS? 02
236 GTO 09
237 RCL 29
238 2
239 *
240 SQRT
241 RCL 04
242 /
243 CHS
244 1
245 +
246 STO 33
247 GTO "VI"
248♦LBL 09
249 RCL 30
250 2
251 *
252 SQRT
253 RCL 09
254 /
255 CHS
256 1
257 +
258 STO 34
259♦LBL "VI"
260 FS? 02
261 RCL 40
262 FC? 02
263 RCL 11
264 RCL 22
265 /
266 FS? 02
267 RCL 24
268 FC? 02
269 RCL 23
270 /
271 2
272 /
273 STO 58
274 RCL 18
275 X↑2
276 RCL 58
277 /
278 2
279 /
280 STO 59
281 X↑2
282 1
283 +
284 SQRT
285 RCL 59
286 -
287 SQRT
288 RCL 58
289 SQRT
290 *
291 FS? 02
292 STO 32
293 FC? 02
294 STO 31
295 FS? 02
296 GTO 11
297 FC? 02
298 GTO "PI"
299♦LBL 11
300 RCL 18
301 67.6
302 X>Y?
303 GTO "PI"
304 RCL 40
305 RCL 22
306 /
307 RCL 24
308 /
309 RCL 18
310 /
311 2
312 /
313 STO 32
314♦LBL "PI"
315 FS? 02
316 RCL 32
317 FC? 02
318 RCL 31
319 FS? 02
320 RCL 40
321 FC? 02
322 RCL 11
323 *
324 FS? 02
325 RCL 34
326 FC? 02
327 RCL 33
328 /
329 FS? 02
330 1
331 FC? 02
332 RCL 45
333 *
334 550
335 /
336 FS? 02
337 STO 41
338 FC? 02
339 STO 35
340♦LBL "PO"
341 RCL 01
342 RCL 04
343 *
344 RCL 00
345 /
346 PI
347 /
348 STO 15
349 FS? 02
350 RCL 28
```

POWER

```
351 FC? 02
352 RCL 27
353 3
354 Y↑X
355 FS? 02
356 RCL 05
357 FC? 02
358 RCL 00
359 [illegible] *
360 [illegible]
361 FS? 02
362 RCL 06
363 FC? 02
364 RCL 01
365 *
366 FS? 02
367 RCL 09
368 FC? 02
369 RCL 04
370 *
371 FS? 02
372 RCL 08
373 FC? 02
374 RCL 03
375 *
376 4400
377 /
378 RCL 22
379 *
380 STO 52
381 RCL 18
382 FS? 02
383 RCL 28
384 FC? 02
385 RCL 27
386 /
387 X↑2
388 4.3
389 *
390 1
391 +
392 RCL 52
393 *
394 FC? 02
395 STO 36
396 FS? 02
397 STO 42
398 FS? 02
399 GTO "PT"
400*LBL "PP"
401 RCL 18
402 3
403 Y↑X
404 RCL 12
405 *
406 RCL 22
407 *
408 1100
409 /
410 STO 37
411*LBL "PC"
412 RCL 19
413 3
414 Y↑X
415 RCL 13
416 *
417 RCL 22
418 *
419 2
420 /
421 550
422 /
423 ST+ 37
424 RCL 19
425 RCL 11
426 *
427 2
428 /
429 550
430 /
431 ST- 35
432 RCL 19
433 RCL 11
434 *
435 550
436 /
437 STO 38
438 RCL 35
439 +
440 RCL 36
441 +
442 RCL 37
443 +
444 STO 39
445*LBL "THRUST"
446 RCL 39
447 550
448 *
449 RCL 02
450 /
451 RCL 10
452 X=0?
453 GTO "MN"
454 /
455 STO 40
456 SF 02
457 GTO "CT"
458*LBL "PT"
459 RCL 41
460 RCL 42
461 +
462 STO 43
463 RCL 39
464 +
465 STO 44
466 FS? 03
467 GTO 13
468 "PT(AC)="
469 ARCL X
470 AVIEW
471 STOP
472*LBL "MN"
473 RCL 39
474 "PT(MR)="
475 ARCL X
476 AVIEW
477 STOP
478 "HI SPD?"
479 PROMPT
480 X=0?
481 GTO "VV"
482 GTO "HSE"
483*LBL 13
484 END
```

VE

Maximum Endurance Velocity

Introduction: This program finds the minimum power required and thus the velocity and power required for maximum endurance. It works with POWER which must be loaded with the geometric and flight parameters (up through and including step 19 in the step-by-step instructions). Upper and lower velocity bounds and the velocity stpe increment are input and VE outputs the velocity and power required for maximum endurance. VE utilizes the same storage registers as POWER and additionally those listed below. No new equations are employed.

Additional Programs Required: POWER

Additional Storage Registers:

Storage Register	Quantity Stored
48	Velocity increment (kts)
50	V-STOP - Stopping velocity (kts)
51	scratch

INSTRUCTIONS	INPUT	FUNCTION	SIZE 060 DISPLAY
1. Initialize program		XEQ VE	*LOAD POWER*
2. Reminder flashes			V-START=?
3. Input lower bound velocity (kts)	50	R/S	INCR=?
4. Input increment (kts)	10	R/S	V-STOP=?
5. Input upper bound Velocity (kts)	150	R/S	V(END)=90
6. Output maximum endurance velocity (kts)		R/S	P(END)=1058

Note: For increased accuracy, after first run, narrow the bounds, decrease the increment and re-run the program.

VE

```
01*LBL "VE"
02 "*LOAD POWER*"
03 AVIEW
04 PSE
05 "*OR PWRC*"
06 AVIEW
07 PSE
08 SF 03
09 "V-START=?"
10 PROMPT
11 1.68889
12 *
13 STO 18
14 0
15 STO 19
16 CF 02
17 XEQ "CT"
18 "INCR=?"
19 PROMPT
20 1.68889
21 *
22 STO 48
23 "V-STOP=?"
24 PROMPT
25 1.68889
26 *
27 STO 50
28*LBL 12
29 RCL 44
30 STO 51
31 RCL 48
32 ST+ 18
33 CF 02
34 XEQ "CT"
35 RCL 51
36 RCL 44
37 X<=Y?
38 GTO 14
39 GTO 15
40*LBL 14
41 RCL 50
42 RCL 18
43 X<=Y?
44 GTO 12
45*LBL 15
46 CF 03
47 RCL 18
48 RCL 48
49 -
50 1.68889
51 /
52 FIX 0
53 "V(END)="
54 ARCL X
55 AVIEW
56 STOP
57 RCL 51
58 "P(END)="
59 ARCL X
60 AVIEW
61 END
```

VMR

Maximum Range Velocity

Introduction: This program determines the maximum range velocity of a given configuration helicopter. The user loads POWER with the basic geometric and flight parameters (up through and including step 19 in the POWER step-by-step instructions). Lower and upper velocity bounds are input by the user, as well as the velocity increment for getting from the lower and upper bound. The phantom shaft horsepower of the aircraft is also entered. This can be determined by first running FUEL. VMR then iterates through POWER in order to determine the minimum ratio of the sum of total power and phantom power <PT(AC) + PSHP> to VF. This is the point of tangency of a line drawn from the point of zero velocity to the power versus velocity curve. This point of tangency is the point of maximum range for the aircraft. VMR utilizes the same storage registers as POWER, and additionally those listed below,

Additional Programs Required: POWER

Equations:

$\text{Tan}\,\phi = \langle PT(AC) + PSHP \rangle / VF$ Definition of $\tan\phi$

Additional Storage Registers:

Storage Register	Stored Quantity
48	Velocity increment (kt)
49	PSHP - Phantom shaft horsepower (SHP)
50	V-STOP - stopping velocity (kt)
54	scratch

VMR

				SIZE 060
	INSTRUCTION	INPUT	FUNCTION	DISPLAY
1.	Initialize program		XEQ VMR	
2.	Reminder flashes			*LOAD POWER*
3.	Reminder flashes			PSHP=?
4.	Input phantom shaft horsepower (SHP)	300	R/S	V-START=?
5.	Input lower bound velocity (kt)	50	R/S	INCR=?
6.	Input increment (kt)	10	R/S	V-STOP=?
7.	Input upper bound velocity (kt)	160	R/S	VMR= 140
8.	Output max range velocity (kt)		R/S	P(VMR)= 1379
9.	Output max range power (SHP)			

NOTE: To increase accuracy, repeat procedure with 10 kt between the lower and upper bounds and an increment of 1 kt.

VMR

```
01♦LBL "VMR"
02 "♦LOAD POWER♦"
03 AVIEW
04 PSE
05 CF 02
06 SF 03
07 "PSHP=?"
08 PROMPT
09 STO 49
10 "V-START=?"
11 PROMPT
12 1.68889
13 *
14 STO 18
15 0
16 STO 19
17 XEQ "CT"
18 RCL 44
19 RCL 49
20 +
21 RCL 18
22 /
23 STO 51
24 "INCR=?"
25 PROMPT
26 1.68889
27 *
28 STO 48
29 "V-STOP=?"
30 PROMPT
31 1.68889
32 *
33 STO 50
34♦LBL 01
35 RCL 48
36 ST+ 18
37 CF 02
38 XEQ "CT"
39 RCL 44
40 RCL 49
41 +
42 RCL 18
43 /
44 RCL 51
45 X>Y?
46 GTO 02
47 GTO 03
48♦LBL 02
49 RCL Y
50 STO 51
51 RCL 50
52 RCL 18
53 X<=Y?
54 GTO 01
55♦LBL 03
56 RCL 18
57 RCL 48
58 -
59 STO 18
60 1.68889
61 /
62 FIX 0
63 "VMR="
64 ARCL X
65 AVIEW
66 STOP
67 CF 02
68 XEQ "CT"
69 RCL 44
70 "P(VMR)="
71 ARCL X
72 AVIEW
73 STOP
74 END
```

WT
(WT LT - Light)
(WT MED - Medium)
(WT HV - Heavy)
Helicopter Weight Estimation

Introduction: These programs are designed to provide by an iterative process, weight estimations to be used in determining final helicopter design weight. Curve fit equations determine the majority of component weight values, such as tail structure and landing gear systems. Equations are listed with the assignment of non-standard storage registers for user modification, if desired. The equations are from ref. 2, Page 20 f. The user must enter into storage the initial values defined in Storage Registers 01 - blade radius (ft); 03 - blade chord (ft); 05 - rotational velocity (rad/sec); 06 - empty weight (lbs); 07 - number of people; 09 - person weight (lbs); 09 - cargo weight (lns); 33 - fuel weight (lbs); 34 - total power (SHP); 36 - number of engines; and 37 - engine weight and/or transmission weight (lbs), if specified.

The program will determine revised empty weight, gross weight and total power requirement based on weight computations plus a percentage change based on prior value. (a 10% or less change is usually desired). The program allows the user to input specific transmission and/or engine weights, or it will generate ts own curve fit values. In addition, the user may choose to retain ininital inputted values or to revise inputs during subsequent iterations of the program.

The program defines useful load as the sum of both number of people (times the person weight specified) and the cargo weight.

All three programs utilize very similar storage registers and prompt similar requests. Program WT MED is demonstrated in the following example.

WT MED (Illustrated)

			Size 060
INSTRUCTION	INPUT	FUNCTION	DISPLAY
1. Initialize Program		XEQ WT MED	R=?
2. Input Radius (ft)	25	R/S	c=?
3. Input chord (ft)	1.9	R/S	b=?
4. Input blade number	4	R/S	RV=?
5. Input RV (rad/sec)	28.8	R/S	We=?
6. Input empty Weight (lbs)	9000	R/S	PEOPLE=?
7. Input People	3	R/S	PERSON WT=?
8. Input Person Wt (lbs)	250	R/S	CARGO=?
9. Input cargo wt (lbs)	6000	R/S	FUEL(lb)=?
10. Input fuel weight (lbs)	3500	R/S	PT(SHP)=?
11. Input total power (SHP)	1600	R/S	NENG=?
12. Input number of engines	2	R/S	ACTWT or R/S? ENG WT(E) =?
Given the option, User choses to input actual weight (ACTWT) specified			
13a. Input specified Engine weight	750	R/S	ACWT or R/S=? TRAN WT(E)=?
14a. Input specified Transmission weight	1600	R/S	REV We= 9276.7

INSTRUCTION	INPUT	FUNCTION	DISPLAY
15. Determine Revised Total Weight (lbs)		R/S	WG=19526.
16. Determine Revised PT (SHP)		R/S	REV PT= 2057.1
17. Determine % Error		R/S	ER(%)= -9.83
13b. Input ENG WT (Accepting Curve fit value)		R/S	ACTWT or R/S? TRANS WT =?
Having accepted Curve fit value for Engine weight, User accepts Curve fit value for Transmission weight			
14b. Accept Transmission weight and output revised Empty weight		R/S	REV We= 8509.4
If user accepted curvefit values for Steps 13b, and 14b, Resultant values for Steps 14,15,16,17 would be: REV We= 8509.4, REV WG= 18759.4, REV PT= 1959.4, and ER(%)= -6.14 User accepts these values or seeks additional iteration with new empty weight, total weight, and total power (results from Steps 14,15,16).			
18. Seeks additional iteration		R/S	ENG WT(E) =?
19. See Step 12 above and repeat or			
20. Re-initialize program		XEQ WT MED	R=?

Weight Estimating Relationships
Storage Register Utilization
Light Helicopter

Storage Register	Stored Quantity
01	We - Empty weight (lbs)
03	People - Number of passengers and crew
05	Cargo - Weight of cargo (lbs)
06	Fuel (lbs) - Fuel weight (lbs)
07	PT (SHP) - Total power (SHP)
08	Neng - Number of engines
09	Gross weight (lbs) $W_g = 1.73 \cdot W_e^{.378}$
10	Total tail surface area (sq ft) $Stt = 0.264 \cdot e^{(.0135 H_p)}$
11	Body surface area (sq ft) $Sb = 194.274 \cdot \ln(W_g) - 1306.779$
12	Main rotor system weight (lbs) $W1 = 408.562 \cdot \ln(S) - 1142.917$
13	Tail rotor system weight (lbs) $W2A = 2.219 \cdot e^{(.0005 W_g)}$
	Tail rotor structure weight (lbs) $W2B = 19.131 \cdot \ln(Stt) - 32.414$
14	Body weight (lbs) $W3 = 0.00901 \cdot Sb^{1.917}$
15	Landing gear weight (lbs) $W4 = -0.0539 \cdot W_g + 200.912$
16	Nacelle weight (lbs) $W5 = 34.0$
17	Propulsion Engine weight (lbs) $W6A = -0.0896 \cdot HP + 221.338$

LIGHT HELICOPTER

Storage Register	Stored Quantity
18	Drive system weight (lbs) $W6B = 17.190 \cdot e^{(.0008 W_g)}$
19	Fuel tanks weight (lbs) $W6C = 0.384 \cdot (Fuel/6.5)^{1.0710}$
20	Flight controls weight (lbs) $W7 = 0.000000000128 \cdot W_g^{3.469}$
21	Auxiliary power system weight (lbs) $W8 = 0.0$
22	Flight instruments weight (lbs) $W9 = 24.571 \cdot e^{(.0004 HP)}$
23	Hydraulics system weight (lbs) $W10 = 0.0$
24	Electrical system weight (lbs) $W11 = -51.0661 \cdot \ln(S_b) + 367.947$
25	Avionics system weight (lbs) $W12 = 105.0$ + Special (if any)
26	Furnishings weight (lbs) $W13 = 19.8\ e^{(.372 People)} + e^{(-.033 S_b)}$
27	Air and Ice system weight (lbs) $W14 = -22.371 \cdot \ln(Sb) + 143.396$
28	Load and Handling equipment weight (lbs) $W15 = 0.0$
33	R - Main rotor radius (ft)
34	c - Main rotor chord (ft)
35	b - Number of blades
36	Person Wt - Weight of an individual (lbs)
37	RV - Main rotor rotational velocity (rad/sec)

LIGHT HELICOPTER

Storage Register	Stored Quantity
38	Scratch
39	Scratch
40	Scratch

Weight Estimating Relationships
Storage Register Utilization
Medium Helicopter

Register	Stored Quantity
01	We - Empty weight (lbs)
03	People - Number of passengers and crew
05	Cargo - Weight of cargo (lbs)
06	Fuel (lbs) - Fuel weight (lbs)
07	PT (SHP) - Total power (SHP)
08	Neng - Number of engines
09	Gross weight (lbs) $W_g = 16239.43 \cdot \ln(W_e) - 130252.76$
10	Total tail surface area (sq ft) $Stt = 0.0376 \cdot H_p - 8.106$
11	Body surface area (sq ft) $S_b = 636.081 \cdot e^{(.000011\ W_g)}$
12	Main rotor system weight (lbs) $W1 = 11.0702 \cdot S - 168.888$
13	Tail rotor system weight (lbs) $W2A = 0.00438 \cdot W_g + 12.470$
	Tail rotor structure weight (lbs) $W2B = 2.411 \cdot Stt - 19.531$
14	Body weight (lbs) $W3 = 0.282 \cdot S_b^{1.272}$
15a or	Landing gear weight (lbs)- Wg less than 6000 lbs $W4 = 0.015 \cdot e^{(.000062 W_g + 8.020)}$
15b	Landing gear weight - Wg greater than 6000 lbs $W4 = 301.577 \cdot \ln(Wg) - 2319.890$
16	Nacelle weight (lbs) $W5 = 0.02 \cdot e^{(.000062 W_g + 8.02)}$

MEDIUM HELICOPTER

Storage Register	Stored Quantity
17	Propulsion Engine weight (lbs)- One engine W6A = 130.0 + 0.451·HP
	Two or more engines W6A = 295.0 + 0.188·HP
18	Drive system weight (lbs) W6B = 741.460·ln(HP) - 4542.042
19	Fuel tanks weight (lbs) W6C = 363.24·ln(Fuel/6.5) - 1656.521
20	Flight controls weight (lbs) $W7 = 210.858 \cdot e^{(.000059\ W_g)}$
21	Auxiliary power system weight (lbs)- One engine W8 = 0.0
	Two Or more engines W8 = 190.0
22	Flight instruments weight (lbs) W9 = 56.0975·ln(HP) - 312.237
23	Hydraulics system weight (lbs) $W10 = 0.00362 \cdot W_g + 11.553$
24	Electrical system weight (lbs) W11 = 481.735·ln(Sb) - 2794.530
25	Avionics system weight (lbs) W12 = 250 + Special (if any)
26	Furnishings weight (lbs) $W13 = 0.175 \cdot S_b + 22.0 \cdot$People - 10.0
27	Air and Ice system weight (lbs) $W14 = 122.458 \cdot \ln(S_b) - 730.252$
28	Load and Handling equipment weight (lbs) W15 = 84.5
33	R - Main rotor radius (ft)

MEDIUM HELICOPTER

Storage Register	Stored Quantity
35	b - Number of blades
36	Person Wt - Weight of an individual (lbs)
37	RV - Main rotor rotational velocity (rad/sec)
38	Scratch
39	Scratch
40	Scratch

Weight Estimating Relationships
Storage Register Utilization
Heavy Helicopter

Storage Register	Stored Quantity
01	We - Empty weight (lbs)
03	People - Number of passengers and crew
05	Cargo - Weight of cargo (lbs)
06	Fuel (lbs) - Fuel weight (lbs)
07	PT (SHP) - Total power (SHP)
08	Neng - Number of engines
09	Gross weight (lbs) $W_g = 4.975 \cdot W_e^{.887}$
10	Total tail surface area (sq ft) $Stt = 60.127 \cdot e^{(.000145\ HP)}$
11	Body surface area (sq ft) $Sb = 426.378 \cdot e^{(.000045\ W_g)}$
12	Main rotor system weight (lbs) $W1 = 707.174 \cdot e^{(.00539\ S)}$
13	Tail rotor system weight (lbs) $W2A = 324.550 \cdot \ln(W_g) - 3021.510$
	Tail rotor structure weight (lbs) $W2B = -18.0 + 2.830\ Stt$
14	Body weight (lbs) $W3 = 2.9818 \cdot Sb - 1321.921$
15	Landing gear weight (lbs) $W4 = 258.358 \cdot e^{(.000041\ W_g)}$
16	Nacelle weight (lbs) $W5 = 0.014 \cdot (0.241\ W_g)^{1.136}$
17	Propulsion Engine weight (lbs)- One engine $W6A = 348.0 + 0.910 \cdot HP$

HEAVY HELICOPTER

Register	Stored Quantity
18	Drive system weight (lbs) $W6B = 0.999 \cdot HP^{.959}$
19	Fuel tanks weight (lbs) $W6C = 454.619 \cdot (Fuel/6.5)^{-.0566}$
20	Flight controls weight (lbs) $W7 = 0.0034 \cdot W_g^{1.224}$
21	Auxiliary power system weight (lbs) $W8 = 139.0$
22	Flight instruments weight (lbs) $W9 = 68.266 \cdot \ln(HP) - 387.598$
23	Hydraulics system weight (lbs) $W10 = 0.000000663 \cdot W_g^{1.863}$
24	Electrical system weight (lbs) $W11 = 9.780 \cdot Sb^{.539}$
25	Avionics system weight (lbs) W12 = 325 + Special (if any)
26	Furnishings weight (lbs) $W13 = 0.159 \cdot Sb + 18.11 \cdot People$
27	Air and Ice system weight (lbs) $W14 = 117.771 \cdot \ln(sb) - 710.594$
28	Load and Handling equipment weight (lbs) $W15 = -72.0 + (0.111 \cdot Sb) + (3.49 \cdot People)$

HEAVY HELICOPTER

Storage Register	Stored Quantity
33	R - Main rotor radius (ft)
34	c - Main rotor chord (ft)
35	b - Number of blades
36	Person Wt - Weight of an individual (lbs)
37	RV - Main rotor rotational velocity (rad/sec)
38	Scratch
39	Scratch
40	Scratch

WT LT

```
01♦LBL "WT LT"
02 RCL 33
03 "R=?"
04 PROMPT
05 STO 33
06 RCL 34
07 "c=?"
08 PROMPT
09 STO 34
10 *
11 RCL 35
12 "b=?"
13 PROMPT
14 STO 35
15 "S"
16 RCL 34
17 *
18 RCL 33
19 *
20 STO 02
21 RCL 37
22 "RV=?"
23 PROMPT
24 STO 37
25 RCL 01
26 "We=?"
27 PROMPT
28 STO 01
29 RCL 03
30 "PEOPLE=?"
31 PROMPT
32 STO 03
33 RCL 36
34 "PERSON WT=?"
35 PROMPT
36 STO 36
37 RCL 03
38 *
39 STO 04
40 RCL 05
41 "CARGO=?"
42 PROMPT
43 STO 05
44 RCL 06
45 "FUEL (LB)=?"
46 PROMPT
47 STO 06
48 RCL 07
49 "PT (SHP)=?"
50 PROMPT
51 STO 07
52 RCL 08
53 "NENG=?"
54 PROMPT
55 STO 08
56♦LBL "WG"
57 RCL 01
58 .378
59 Y↑X
60 173.701
61 *
62 STO 09
63♦LBL "STT"
64 RCL 07
65 .0135
66 *
67 E↑X
68 .264
69 *
70 STO 10
71♦LBL "Sb"
72 RCL 09
73 LN
74 194.274
75 *
76 1306.779
77 -
78 STO 11
79♦LBL "W1"
80 RCL 02
81 LN
82 408.562
83 *
84 1142.917
85 -
86 STO 12
87 STO 29
88♦LBL "W2A"
89 RCL 09
90 .0005
91 *
92 E↑X
93 2.219
94 *
95♦LBL "W2B"
96 32.414
97 -
98 RCL 10
99 LN
100 19.131
101 *
102 +
103 STO 13
104 ST+ 29
105♦LBL "W3"
106 RCL 11
107 1.917
108 Y↑X
109 .00901
110 *
111 STO 14
112 ST+ 29
113♦LBL "W4"
114 RCL 09
115 -.0539
116 *
117 200.912
118 +
119 STO 15
120 ST+ 29
121♦LBL "W5"
122 34
123 STO 16
124 ST+ 29
125♦LBL "W6A"
126 RCL 07
127 -.0896
128 *
129 221.388
130 +
131 STO 17
132 CF 21
133 "ACTWT OR R/S"
134 AVIEW
135 PSE
136 SF 21
137 "ENG WT(Σ)=?"
138 PROMPT
139 STO 17
140 ST+ 29
141♦LBL "W6B"
142 RCL 09
143 .0008
144 *
145 E↑X
146 17.198
147 *
148 STO 18
149 CF 21
150 "ACTWT OR R/S"
151 AVIEW
152 PSE
153 SF 21
154 "TRAN WT(Σ)=?"
155 PROMPT
156 STO 18
157 ST+ 29
158♦LBL "W6C"
159 RCL 06
160 6.5
161 /
162 1.071
163 Y↑X
164 .384
165 *
166 STO 19
167 ST+ 29
168♦LBL "W7"
169 RCL 09
170 3.469
171 Y↑X
172 1.281 E-10
173 *
174 STO 20
175 ST+ 29
176♦LBL "W8"
177 0
178 STO 21
179 ST+ 29
180♦LBL "W9"
181 RCL 07
182 .0004
183 *
184 E↑X
185 24.571
186 *
187 STO 22
188 ST+ 29
189♦LBL "W10"
190 0
191 STO 23
192 ST+ 29
193♦LBL "W11"
194 RCL 11
195 LN
196 -51.0661
197 *
198 367.947
199 +
200 STO 24
```

WT LT

```
201 ST+ 29
202*LBL "W12"
203 RCL 07
204 .003
205 *
206 E↑X
207 1120.354
208 *
209 LN
210 -122.282
211 *
212 1062.004
213 +
214 STO 25
215 ST+ 29
216*LBL "W13"
217 RCL 03
218 .372
219 *
220 E↑X
221 19.8
222 *
223 RCL 11
224 -.033
225 *
226 E↑X
227 +
228 STO 26
229 ST+ 29
230*LBL "W14"
231 RCL 11
232 LN
233 -22.371
234 *
235 143.396
236 +
237 STO 27
238 ST+ 29
239*LBL "W15"
240 0
241 STO 28
242 "NEW We"
243 ST+ 29
244 "REV We="
245 ARCL 29
246 AVIEW
247 RCL 29
248 STO 01
249 "FUEL WT"
250 RCL 06
251 ST+ 29
252 "CARGO"
253 RCL 05
254 ST+ 29
255 "PEOPLE WT"
256 RCL 04
257 "NEW WT"
258 ST+ 29
259 "WG="
260 ARCL 29
261 AVIEW
262 XEQ "PT"
263 "REV PT="
264 ARCL 07
265 AVIEW
266 ADV
267 RCL 09
268 RCL 29
269 -
270 RCL 29
271 /
272 100
273 *
274 "ER(%)="
275 ARCL X
276 PROMPT
277 RCL 29
278 STO 09
279 GTO "W1"
280*LBL "PT"
281 .125
282 RCL 35
283 *
284 RCL 34
285 *
286 "CdO"
287 .01
288 *
289 "DENSITY"
290 .0023769
291 *
292 RCL 37
293 3
294 Y↑X
295 *
296 RCL 33
297 4
298 Y↑X
299 *
300 STO 07
301 2
302 PI
303 *
304 "DENSITY"
305 .0023769
306 *
307 RCL 33
308 X↑2
309 *
310 SQRT
311 "B"
312 .97
313 *
314 1/X
315 RCL 29
316 1.5
317 Y↑X
318 *
319 RCL 07
320 +
321 550
322 /
323 STO 07
324 RTN
325 END
```

WT MED

```
01♦LBL "WT MED"
02 RCL 33
03 "P=?"
04 PROMPT
05 STO 33
06 RCL 34
07 "c=?"
08 PROMPT
09 STO 34
10 *
11 RCL 35
12 "b=?"
13 PROMPT
14 STO 35
15 "S"
16 RCL 34
17 *
18 RCL 33
19 *
20 STO 02
21 RCL 37
22 "RV=?"
23 PROMPT
24 STO 37
25 RCL 01
26 "We=?"
27 PROMPT
28 STO 01
29 RCL 03
30 "PEOPLE=?"
31 PROMPT
32 STO 03
33 RCL 36
34 "PERSON WT=?"
35 PROMPT
36 STO 36
37 RCL 03
38 *
39 STO 04
40 RCL 05
41 "CARGO=?"
42 PPOMPT
43 STO 05
44 RCL 06
45 "FUEL (LB)=?"
46 PROMPT
47 STO 06
48 RCL 07
49 "PT (SHP)=?"
50 PROMPT
51 STO 07
52 RCL 08
53 "NENG=?"
54 PROMPT
55 STO 08
56♦LBL "WG"
57 RCL 01
58 LN
59 16239.43
60 *
61 130252.76
62 -
63 STO 09
64♦LBL "STT"
65 RCL 07
66 .0376
67 *
68 8.106
69 -
70 STO 10
71♦LBL "Sb"
72 RCL 09
73 .000011
74 *
75 E↑X
76 636.081
77 *
78 STO 11
79♦LBL "W1"
80 RCL 02
81 11.0702
82 *
83 168.888
84 -
85 STO 12
86 STO 29
87♦LBL "W2A"
88 RCL 09
89 .00438
90 *
91 12.47
92 +
93♦LBL "W2B"
94 RCL 10
95 2.411
96 *
97 +
98 19.531
99 -
100 STO 13
101 ST+ 29
102♦LBL "W3"
103 RCL 11
104 1.272
105 Y↑X
106 .282
107 *
108 STO 14
109 ST+ 29
110 RCL 09
111 6000
112 X>Y?
113 GTO "W41"
114♦LBL "W42"
115 RCL 09
116 LN
117 301.577
118 *
119 2319.89
120 -
121 STO 15
122 ST+ 29
123 GTO "W5"
124♦LBL "W41"
125 RCL 09
126 .000062
127 *
128 8.82
129 +
130 E↑X
131 .025
132 *
133 STO 15
134 ST+ 29
135♦LBL "W5"
136 RCL 09
137 .000062
138 *
139 8.82
140 +
141 E↑X
142 .82
143 *
144 STO 16
145 ST+ 29
146 1
147 RCL 08
148 X>Y?
149 GTO "W6A2"
150♦LBL "W6A1"
151 RCL 07
152 .24
153 *
154 130
155 +
156 STO 17
157 CF 21
158 "ACTWT OR R/S"
159 AVIEW
160 PSE
161 SF 21
162 "ENG WT(Σ)=?"
163 PROMPT
164 STO 17
165 ST+ 29
166 GTO "W6B"
167♦LBL "W6A2"
168 RCL 07
169 .19
170 *
171 350.4
172 +
173 STO 17
174 CF 21
175 "ACTWT OR R/S"
176 AVIEW
177 PSE
178 SF 21
179 "ENG WT(Σ)=?"
180 PROMPT
181 STO 17
182 ST+ 29
183♦LBL "W6B"
184 RCL 07
185 LN
186 741.460
187 *
188 4542.842
189 -
190 STO 18
191 CF 21
192 "ACTWT OR R/S"
193 AVIEW
194 PSE
195 SF 21
196 "TRAN WT(Σ)=?"
197 PROMPT
198 STO 18
199 ST+ 29
200♦LBL "W6C"
```

WT MED

```
201 RCL 06
202 6.5
203 /
204 LN
205 363.24
206 *
207 1656.521
208 -
209 STO 19
210 ST+ 29
211*LBL "W7"
212 RCL 09
213 .000059
214 *
215 E↑X
216 210.858
217 *
218 STO 20
219 ST+ 29
220*LBL "W8"
221 0.
222 STO 21
223 ST+ 29
224 1
225 RCL 08
226 X<=Y?
227 GTO "W9"
228 190
229 STO 21
230 ST+ 29
231*LBL "W9"
232 RCL 07
233 LN
234 56.0975
235 *
236 312.237
237 -
238 STO 22
239 ST+ 29
240*LBL "W10"
241 RCL 09
242 .00362
243 *
244 11.553
245 +
246 STO 23
247 ST+ 29
248*LBL "W11"
249 RCL 11
250 LN
251 481.735
252 *
253 2794.53
254 -
255 STO 24
256 ST+ 29
257*LBL "W12"
258 RCL 07
259 .139
260 *
261 77.823
262 +
263 STO 25
264 ST+ 29
265*LBL "W13"
266 RCL 03
267 22
268 *
269 10
270 -
271 RCL 11
272 .175
273 *
274 +
275 STO 26
276 ST+ 29
277*LBL "W14"
278 RCL 11
279 LN
280 122.458
281 *
282 730.252
283 -
284 STO 27
285 ST+ 29
286*LBL "W15"
287 84.5
288 STO 28
289 "NEW We"
290 ST+ 29
291 "REV We="
292 ARCL 29
293 AVIEW
294 RCL 29
295 STO 01
296 "FUEL WT"
297 RCL 06
298 ST+ 29
299 "CARGO"
300 RCL 05
301 ST+ 29
302 "PEOPLE WT"
303 RCL 04
304 "NEW WG"
305 ST+ 29
306 "WG="
307 ARCL 29
308 AVIEW
309 XEQ "PT"
310 "REV PT="
311 ARCL 07
312 AVIEW
313 ADV
314 RCL 09
315 RCL 29
316 -
317 RCL 29
318 /
319 100
320 *
321 "ER(%)="
322 ARCL X
323 PROMPT
324 RCL 29
325 STO 09
326 GTO "W1"
327*LBL "PT"
328 .125
329 RCL 35
330 *
331 RCL 34
332 *
333 "CdO"
334 .01
335 *
336 "DENSITY"
337 .0023769
338 *
339 RCL 37
340 3
341 Y↑X
342 *
343 RCL 33
344 4
345 Y↑X
346 *
347 STO 07
348 2
349 PI
350 *
351 "DENSITY"
352 .0023769
353 *
354 RCL 33
355 X↑2
356 *
357 SQRT
358 "B"
359 .97
360 *
361 1/X
362 RCL 29
363 1.5
364 Y↑X
365 *
366 RCL 07
367 +
368 550
369 /
370 STO 07
371 RTN
372 END
```

WT HV

```
01♦LBL "WT HV"
02 RCL 33
03 "R=?"
04 PROMPT
05 STO 33
06 RCL 34
07 "c=?"
08 PROMPT
09 STO 34
10 *
11 RCL 35
12 "b=?"
13 PROMPT
14 STO 35
15 "S"
16 RCL 34
17 *
18 RCL 33
19 *
20 STO 02
21 RCL 37
22 "PV=?"
23 PROMPT
24 STO 37
25 RCL 01
26 "We=?"
27 PROMPT
28 STO 01
29 RCL 03
30 "PEOPLE=?"
31 PROMPT
32 STO 03
33 RCL 36
34 "PERSON WT=?"
35 PROMPT
36 STO 36
37 RCL 03
38 *
39 STO 04
40 RCL 05
41 "CARGO=?"
42 PROMPT
43 STO 05
44 RCL 06
45 "FUEL (LB)=?"
46 PROMPT
47 STO 06
48 RCL 07
49 "PT (SHP)=?"
50 PROMPT
51 STO 07
52 RCL 08
53 "NENG=?"
54 PROMPT
55 STO 08
56 RCL 31
57 "TANDEM?"
58 PROMPT
59 STO 31
60♦LBL "WG"
61 RCL 01
62 .887
63 Y↑X
64 4.975
65 *
66 STO 09
67♦LBL "STT"
68 RCL 07
69 .000145
70 *
71 E↑X
72 60.127
73 *
74 STO 10
75♦LBL "Sb"
76 RCL 09
77 .000045
78 *
79 E↑X
80 426.378
81 *
82 STO 11
83♦LBL "W1"
84 RCL 02
85 .00539
86 *
87 E↑X
88 707.174
89 *
90 STO 12
91 STO 29
92♦LBL "W6A"
93 RCL 07
94 .91
95 *
96 348
97 +
98 STO 17
99 CF 21
100 "ACTWT OR R/S"
101 AVIEW
102 PSE
103 SF 21
104 "ENG WT(Σ)=?"
105 PROMPT
106 STO 17
107 ST+ 29
108♦LBL "W2A"
109 RCL 09
110 LN
111 324.55
112 *
113 3021.51
114 -
115♦LBL "W2B"
116 RCL 10
117 2.83
118 *
119 +
120 18.0
121 -
122 STO 13
123 ST+ 29
124♦LBL "W3"
125 RCL 11
126 2.9818
127 *
128 1321.921
129 -
130 STO 14
131 ST+ 29
132 RCL 31
133 X<=0?
134 GTO "W4"
135 0
136 STO 10
137 STO 13
138 RCL 09
139 .000041
140 *
141 E↑X
142 567.688
143 *
144 STO 11
145 RCL 12
146 2
147 *
148 STO 12
149 STO 29
150 RCL 07
151 .000198
152 *
153 E↑X
154 565.507
155 *
156 STO 17
157 ST+ 29
158 RCL 11
159 LN
160 3467.291
161 *
162 22118.298
163 -
164 STO 14
165 ST+ 29
166♦LBL "W4"
167 RCL 09
168 .000041
169 *
170 E↑X
171 258.358
172 *
173 STO 15
174 ST+ 29
175♦LBL "W5"
176 RCL 09
177 .2041
178 *
179 1.136
180 Y↑X
181 .014
182 *
183 STO 16
184 ST+ 29
185♦LBL "W6B"
186 RCL 07
187 .959
188 Y↑X
189 .999
190 *
191 STO 18
192 CF 21
193 "ACTWT OR R/S"
194 AVIEW
195 PSE
196 SF 21
197 "TRAN WT(Σ)=?"
198 PROMPT
199 STO 18
200 ST+ 29
```

WT HV

```
201*LBL "W6C"
202 RCL 06
203 6.5
204 /
205 -.0566
206 Y↑X
207 454.619
208 *
209 STO 19
210 ST+ 29
211*LBL "W7"
212 RCL 09
213 1.224
214 Y↑X
215 .00334
216 *
217 STO 20
218 ST+ 29
219*LBL "W8"
220 139.0
221 STO 21
222 ST+ 29
223*LBL "W9"
224 RCL 07
225 LN
226 68.266
227 *
228 387.598
229 -
230 STO 22
231 ST+ 29
232*LBL "W10"
233 RCL 09
234 1.863
235 Y↑X
236 .000000663
237 *
238 STO 23
239 ST+ 29
240*LBL "W11"
241 RCL 11
242 .539
243 Y↑X
244 9.78
245 *
246 STO 24
247 ST+ 29
248*LBL "W12"
249 RCL 07
250 LN
251 16744.967
252 *
253 108666
254 -
255 .536
256 Y↑X
257 1.9
258 *
259 STO 25
260 ST+ 29
261*LBL "W13"
262 RCL 11
263 .159
264 *
265 RCL 03
266 18.11
267 *
268 +
269 STO 26
270 ST+ 29
271*LBL "W14"
272 RCL 11
273 LN
274 117.771
275 *
276 710.594
277 -
278 STO 27
279 ST+ 29
280*LBL "W15"
281 RCL 03
282 3.49
283 *
284 72
285 -
286 RCL 11
287 .111
288 *
289 +
290 STO 28
291 "NEW We"
292 ST+ 29
293 "REV We="
294 ARCL 29
295 AVIEW
296 RCL 29
297 STO 01
298 "FUEL WT"
299 RCL 06
300 ST+ 29
301 "CARGO"
302 RCL 05
303 ST+ 29
304 "PEOPLE WT"
305 RCL 04
306 "NEW WG"
307 ST+ 29
308 "WG="
309 ARCL 29
310 AVIEW
311 XEQ "PT"
312 "REV PT="
313 ARCL 07
314 AVIEW
315 ADV
316 RCL 09
317 RCL 29
318 -
319 RCL 29
320 /
321 100
322 *
323 "ER(%)="
324 ARCL X
325 PROMPT
326 RCL 29
327 STO 09
328 GTO "W1"
329*LBL "PT"
330 .125
331 RCL 35
332 *
333 RCL 34
334 *
335 "CdO"
336 .01
337 *
338 "DENSITY"
339 .0023769
340 *
341 RCL 37
342 3
343 Y↑X
344 *
345 RCL 33
346 4
347 Y↑X
348 *
349 STO 07
350 2
351 PI
352 *
353 "DENSITY"
354 .0023769
355 *
356 RCL 33
357 X↑2
358 *
359 SQRT
360 "B"
361 .97
362 *
363 1/X
364 RCL 29
365 1.5
366 Y↑X
367 *
368 RCL 07
369 +
370 550
371 /
372 STO 07
373 RTN
374 END
```

END

FILMED

1-86

DTIC